


# 10

## STARTER PROGRAMS


FROM

---

# FAMILY COMPUTING



BY JOEY LATIMER  
FOR  
APPLE, ATARI, COMMODORE 64 and VIC-20, TI, TIMEX and TRS-80



# **TABLE OF CONTENTS**

**1**

INTRODUCTION: STARTING TO PROGRAM

**1**

THE ABC'S OF PROGRAMMING IN BASIC

**3**

APPLE PROGRAMS

**15**

ATARI PROGRAMS

**28**

COMMODORE 64 AND VIC-20 PROGRAMS

**42**

TEXAS INSTRUMENTS PROGRAMS

**57**

TIMEX PROGRAMS

**71**

TRS-80 PROGRAMS

# INTRODUCTION

## STARTING TO PROGRAM

You've probably heard that computers are dumb. In a sense they are—they don't know anything until a human comes along and tells them what to do. Telling a computer what to do is called programming. It's not hard to learn to program. One way is by learning BASIC, one of the many languages used by humans, to talk to computers. The programs in this book are all written in BASIC.

When you program, some keys and parts of your computer keyboard are especially important. Be sure you can locate the quotation mark and the semicolon, and that you know how to space both forward and backward.

The chart below tells you how to make your particular computer do certain things when you program. It also lists some common commands. Commands are the words you key in when you want to tell your computer to do something. It also lists what your computer will tell you to let you know you've made a mistake in your typing. The most important thing to keep in mind when you're talking to your computer is that you must be precise. The computer will try to do exactly what you tell it, so you must be careful to tell it exactly what you want it to do.

## THE ABCs OF PROGRAMMING IN BASIC

This chart tells you what to look for, what to key in, or what something on your screen means.

THE SIGN OR DIRECTION	APPLE	TIMEX	TI	ATARI	COMMODORE	TRS-80
Indication of next symbol	flashing cursor	cursor	flashing cursor	cursor	flashing cursor	flashing cursor
To correct	backspace	delete	backspace	delete/ backspace	insert/ delete	backspace
To enter a command	RETURN	ENTER	ENTER	RETURN	RETURN	ENTER
Mistake in command	SYNTAX ERROR	SYNTAX ERROR	INCORRECT STATEMENT	ERROR	SYNTAX ERROR	SYNTAX ERROR
To clear screen	HOME	CLS	CALL CLEAR	PRINT CHR\$ (125)	PRINT CHR\$ (147)	CLS
To indicate new program	NEW	NEW	NEW	NEW	NEW	NEW
To print something	PRINT	PRINT	PRINT	PRINT	PRINT	PRINT
To see all the commands in a program	LIST	LIST	LIST	LIST	LIST	LIST
To make the program do what you've told it to	RUN	RUN	RUN	RUN	RUN	RUN
To indicate end of program	END	STOP	END	END	END	END
To stop program	CONTROL/ RESET	BREAK	FUNCTION/ 4	BREAK	RUN/STOP	BREAK

## APPLE II PLUS AND APPLE IIe

## Bonus Program #1 CARTOON ROBOT

Riddle: What's sweet, but square; high tech, yet down to earth; and brilliant, with the I.Q. of a doughnut?

Give up?

The answer is: The fabulous KISSING ROBOT.

You'll key in a program using FOR and NEXT to make a cartoon. The command FOR and NEXT are used for counting.

```

14 HOME
16 PRINT "()******()"
18 PRINT "***          ***"
20 PRINT "*** (O)      (O) ***"
22 PRINT "***          ***"
24 PRINT "***      V      ***"
26 PRINT "***          ***"
28 PRINT "***      ---      ***"
30 FOR T = 1 TO 75: NEXT T
32 HOME
34 PRINT "()******()"
36 PRINT "***          ***"
38 PRINT "*** (O)      (-) ***"
40 PRINT "***          ***"
42 PRINT "***      V      ***"
44 PRINT "***          ***"
46 PRINT "***      O      ***"
48 FOR T = 1 TO 75: NEXT T
50 HOME
52 PRINT "()******()"
54 PRINT "***          ***"
56 PRINT "*** (O)      (O) ***"
58 PRINT "***          ***"
60 PRINT "***      V      ***"
62 PRINT "***          ***"
64 PRINT "***      O      ***"
66 FOR T = 1 TO 75: NEXT T
68 GOTO 14

```

This program works like a real cartoon. The robot is printed on the screen, and erased, three times. Each time it is printed, there are small changes made, which give the illusion of movement. Line 68 GOTO 14 starts the entire process over again. The FOR/NEXT commands are used as time delays between pictures. You can change the speed of the cartoon by changing the 75, in the FOR/NEXT lines, to a different number. Decreasing the number will make the cartoon faster.

APPLE

## Bonus Program #2 FAMILY DECISION MAKER

How would you like to use your home computer for solving problems like, "Who will use the computer first, Jimmy or Bobby?"...or how about, "Should we use the t.v. to watch a movie, or play with the computer?". The FAMILY DECISION MAKER can help you solve these problems, and more. It will make the decision for you, by picking a random choice. All you have to do is to type in the options.

```

10 HOME
20 PRINT "FAMILY DECISION MAKER"
30 FOR T = 1 TO 1500:NEXT T
40 HOME
50 PRINT "TYPE IN THE OPTIONS"
60 PRINT "AND THE COMPUTER WILL DECIDE"
70 INPUT "WHAT IS OPTION #1 ";O1$
80 INPUT "WHAT IS OPTION #2 ";O2$
90 PRINT "I'M THINKING IT OVER....."
100 FOR T = 1 TO 3000:NEXT T
110 C=INT(RND(1)*2)+1
120 HOME
130 PRINT "MY CHOICE IS:"
140 IF C = 1 THEN PRINT O1$
150 IF C = 2 THEN PRINT O2$

```

In this program the computer makes its choice in line 110. Lines 90 and 100 are where the computer is, "thinking it over". You probably noticed that the computer isn't really "thinking it over". It's actually counting up to 3000, then executing line 110. The choice is printed on the screen in lines 130-150.

The VARIABLES are:

```

T=time delay
C=choice
O1$=option #1
O2$=option #2

```

APPLE

## Bonus Program #3 REACTION TIMER

Here is a program to test your reaction time. When the computer says "GO!", you must hold down CONTROL and press the RESET key as quickly as you can. Compare your score with the chart in the program. Good luck!

```

10 HOME
20 PRINT "TEST YOUR REACTION TIME"
30 PRINT "AGAINST THE COMPUTER."
40 PRINT "WHEN THE COMPUTER SAYS 'GO!'"
50 PRINT "HOLD CONTROL & PRESS RESET KEY"
60 PRINT "YOUR SCORE IS THE HIGHEST NUMBER YOU SEE"
70 PRINT:PRINT "01-10=LIGHTNIN' 10-20=QUICK!"
80 PRINT "20-30=AVERAGE 30-50=NAPPING"
90 PRINT:PRINT:PRINT "PRESS RETURN"
100 PRINT "WHEN YOU ARE READY"
110 INPUT A$
120 HOME:PRINT "ON YOUR MARK"
130 FOR T = 1 TO 1000:NEXT T:PRINT "GET SET!"
140 FOR T = 1 TO INT(RND(1)*5000):NEXT T
150 HOME:PRINT "GO!"
160 FOR T = 1 TO 50:PRINT T:NEXT T
170 PRINT "SOMEONE WAKE THIS PERSON UP!"

```

The VARIABLE, of the FOR/NEXT statement in line 140, equals a RANDOM INTEGER between one and five thousand. This causes the time delay to be different each time the program is RUN. When you press the RESET key the computer may say, "Break in 160". This is normal for the program. Your score is the highest number you see. Type RUN and press RETURN to play again.

APPLE

## Bonus Program #4 M.P.G. RECORDER

If you're like me, you never take the trouble to figure out your car's miles per gallon (M.P.G.). Even having a calculator handy has never helped, though there are only three basic numbers to calculate. This is one more instance in which wanting to use my computer motivates me to do the fairly simple task I've managed to ignore. My mechanic tells me that I should check my M.P.G. after every five fill-ups. That way, if my M.P.G. starts dropping, I can take my car in for a checkup...before it's too late.

```
10 HOME
20 PRINT "MPG CALCULATOR"
30 FOR T = 1 TO 1500:NEXT T
40 HOME
50 PRINT "THIS IS A PROGRAM TO FIGURE OUT"
60 PRINT "THE MILES PER GALLON YOUR CAR GETS"
70 PRINT "HOW MANY MILES HAVE YOU DRIVEN"
80 PRINT "DURING THE PAST FIVE FILL-UPS"
90 INPUT M
100 PRINT "HOW MANY GALLONS OF GAS DID YOU USE"
110 PRINT "IN THE PAST FIVE FILL-UPS"
120 INPUT G
130 MPG=M/G
140 PRINT "YOU HAVE BEEN GETTING "MPG
150 PRINT "MILES PER GALLON"
```

Notice that we used G as the VARIABLE for gas, M as the VARIABLE for miles, and MPG as the VARIABLE for miles per gallon. In line 130,  $MPG=M/G$  means miles per gallon equals miles divided by gallons.

## APPLE

## Bonus Program #5 COUPON CALCULATOR

Computers are pretty good at solving problems and presenting the results in a manner which is easy to read. This program can be used to display the amount of money you will save with your shopping coupons.

```

10 HOME
20 PRINT "COUPON CALCULATOR"
30 PRINT:PRINT "TO FIND OUT HOW MUCH YOU'LL SAVE"
40 PRINT "WITH YOUR SHOPPING COUPONS"
50 PRINT "ANSWER THE FOLLOWING QUESTIONS"
60 PRINT "THEN PRESS THE RETURN KEY"
70 PRINT:PRINT "HOW MANY COUPONS DO YOU HAVE"
80 INPUT C
90 PRINT:PRINT "ENTER THE AMOUNT OF A COUPON"
100 PRINT "DON'T USE A DOLLAR SIGN"
110 PRINT "DO USE A DECIMAL POINT":PRINT
120 FOR E = 1 TO C: PRINT "ENTER VALUE OF COUPON#"E
130 INPUT A
140 T=T+A
150 NEXT E
160 PRINT:PRINT "$"T" WILL BE SAVED"

```

The VARIABLES in this program are:

C=number of coupons

E=coupon#

A=value of coupons

T=total value of coupons

Line 140 adds up the total, each time a value is entered into the computer. Line 160 prints the total value to be saved.

## APPLE

## Bonus Program #6 SPORTS FORECASTER

The SPORTS FORECASTER can be a handy program if you enjoy sports. This program will take a team's current record and project, based on winning percentage, what the team's record will be at the end of the season.

```

10 HOME
20 PRINT "SPORTS FORECASTER"
30 FOR Z = 1 TO 1500:NEXT Z:PRINT
40 PRINT "THIS PROGRAM WILL FORECAST A TEAM'S"
50 PRINT "FINAL WIN AND LOSS RECORD"
60 PRINT "BASED ON ITS CURRENT RECORD"
70 PRINT:PRINT "ANSWER EACH QUESTION"
80 PRINT "THEN PRESS RETURN"
90 PRINT:PRINT "HOW MANY GAMES DOES THE TEAM PLAY"
100 INPUT T
110 PRINT:PRINT "HOW MANY WINS DO THEY HAVE NOW"
120 INPUT W
130 PRINT "HOW MANY LOSSES DO THEY HAVE NOW"
140 INPUT L
150 P=W/(W+L):Y=T*P:D=T-Y
160 PRINT:PRINT "END OF THE SEASON PROJECTION:"
170 PRINT:PRINT "WINS="INT(Y)" LOSSES="INT(D)+1

```

The VARIABLES are:

Z=time delay variable

T=total games in season

W=games won

L=games lost

P=percentage of games won

Y=end of year games projected won

D=end of year games projected lost

The forecast is completed in line 150 when the winning percentage (P) is established by dividing the number of completed games (W+L) into the total games won so far (W). The total wins for the year is estimated by multiplying the amount of games in the season (T) by the winning percentage (P). The year end losses are determined by subtracting the end of year projected games won (Y) from the total games in the season (T).

## APPLE

## Bonus Program #7 SHOWER MONITOR

Getting into the shower, day after day, and finding cold water can be a drag. I'm sure that large families know what I'm talking about. Bonus Program #7 has been designed to whip, morning bathroom confusion. It's called the SHOWER MONITOR. You type in the names and the computer picks the shower order.

```

10 HOME
20 PRINT "SHOWER MONITOR"
30 FOR T= 1 TO 2000:NEXT T:HOME
40 PRINT "THIS PROGRAM IS DESIGNED TO HELP"
50 PRINT "FAMILIES DECIDE, IN A FAIR WAY"
60 PRINT "THE ORDER IN WHICH THE SHOWER IS USED"
70 PRINT "IN THE MORNING."
80 PRINT:PRINT "EACH PERSON'S NAME IS TYPED INTO THE"
90 PRINT "COMPUTER. THEN THE COMPUTER, RANDOMLY,"
100 PRINT "CHOOSES THE ORDER (AS IF OUT OF A HAT)."
```

110 PRINT "TYPE EACH ANSWER, THEN PRESS RETURN."

120 PRINT:PRINT "HOW MANY PEOPLE IN YOUR FAMILY"

130 INPUT P

140 PRINT:PRINT "TYPE IN THE NAMES, ONE AT A TIME."

150 PRINT "THEN PRESS RETURN."

160 FOR H = 1 TO P

170 INPUT N\$(H)

180 NEXT H

190 HOME:PRINT "THIS IS THE SHOWER ORDER TODAY:"

200 PRINT:FOR R = 1 TO P

210 X=INT(RND(1)\*P)+1

220 IF N\$(X)="" THEN 210

230 PRINT N\$(X)

240 N\$(X)=""

250 NEXT R

260 GOTO 260

The VARIABLES are:

P=number of people in family

H=array parking lot#

X=random number

N\$(X)=name of person X in array

R=counting variable

You are probably wondering what an ARRAY is. An ARRAY is a computer parking lot. In an ARRAY you don't park cars. Rather, you park words and numbers. In this program we parked the name of each person in an ARRAY location (such as N\$(1)="Mom", N\$(2)="Larry", N\$(3)="Rick", etc). The names are loaded into the ARRAY in lines 160-180. The RANDOM shower order is determined in lines 200-250. Can you figure out why a name isn't picked more than once?

## APPLE

## Bonus Program #8 NUMBER GAMES FOR TWO

Computers are great for playing games. They can be programmed to make games, both unpredictable and exciting. Here is a super game for two people. The computer "pulls a number out of its hat", and the players take turns trying to guess the number. The player with the most correct guesses, after seven rounds, is the champ. Switch sides after seven rounds. You will be surprised at the strategies involved.

```

10 HOME
20 PRINT "THIS IS A NUMBER GAME FOR TWO PEOPLE"
30 PRINT "THE COMPUTER PICKS A NUMBER BETWEEN 1 AND 500."
40 PRINT "THE PLAYERS TAKE TURNS GUESSING THE NUMBER"
50 PRINT "UNTIL SOMEONE GUESSES THE NUMBER"
60 PRINT "THE PLAYER GUESSING THE MOST NUMBERS,"
70 PRINT "AFTER 7 ROUNDS, IS THE WINNER"
80 PRINT : INPUT "WHAT IS PLAYER #1'S NAME? ";P1$
90 PRINT : INPUT "WHAT IS PLAYER #2'S NAME? ";P2$
100 R=R+1:N = INT (RND(1) * 500) +1
110 IF R>1 THEN PRINT"THE SCORE IS "P1$="P1" "P2$="P2
120 FOR T = 1 TO 2500:NEXT T
130 IF R > 7 THEN 260
140 FOR T = 1 TO 1000: NEXT T
150 HOME : PRINT "ROUND "R", "P1$"'S TURN"
160 PRINT : INPUT "WHAT IS YOUR GUESS? ";G1
170 IF G1 < N THEN PRINT "TOO LOW, "P1$:GOTO 200
180 IF G1 > N THEN PRINT "TOO HIGH":GOTO 200
190 PRINT "YOU GOT IT "P1$:P1 = P1 + 1:GOTO 100
200 FOR T = 1 TO 1000: NEXT T
210 HOME:PRINT "ROUND "R", "P2$"'S TURN"
220 PRINT:INPUT "WHAT IS YOUR GUESS? ";G2
230 IF G2 < N THEN PRINT "TOO LOW":GOTO 140
240 IF G2 > N THEN PRINT "TOO HIGH":GOTO 140
250 PRINT "YOU GOT IT "P2$:P2 = P2 + 1:GOTO 100
260 HOME:FOR T = 1 TO 1000: NEXT T
270 IF P1 > P2 THEN PRINT P1$ "CREAMED "P2$ "P1" TO "P2:END
280 PRINT P2$ "WASTED "P1$ "P2" TO "P1

```

The VARIABLES are:

P1\$=player #1

P2\$=player #2

P1=player #1 score

P2=player #2 score

G1=player #1 guess

G2=player #2 guess

R=round#

T=time delay variable

N=secret number

The secret number (N) is picked in line 100. To alter the limits of the secret number, you can change the 500 to a larger or smaller number. Try 10000, for instance. In several places you may notice symbols like this: > or this: < . The symbol > means "greater than" and < means "less than". Can you make this game work with four players?

APPLE

## Bonus Program #9 PRACTICAL JOKER

Are you ready for some laughs? If so, Bonus Program #9 is the one for you. It's called the PRACTICAL JOKE PROGRAM. Here's how it works:

You type the program into your computer while the victim isn't around. RUN the program. The computer will ask you questions about the victim. You answer all the questions until the computer says, "PRESS ANY KEY TO START THE JOKE". Press a key and the joke is ready for the victim. When the victim comes back, the computer will seem to know all about him/her. Just say that you're hooked up to the MASTER COMPUTER, and it knows EVERYTHING!!!

```

10 HOME
20 PRINT "PRACTICAL JOKE PROGRAM"
30 FOR T=1 TO 1000:NEXT T
40 HOME
50 INPUT "WHAT IS THE VICTIM'S NAME ";N$
60 INPUT "IS THE VICTIM MALE OR FEMALE ";MF$
70 IF MF$= "MALE" OR MF$= "M" THEN G$= "HE":P$= "HIS"
80 IF MF$= "FEMALE" OR MF$= "F" THEN G$= "SHE":P$= "HER"
90 PRINT "WHAT CITY IS "G$" FROM?":INPUT C$
100 PRINT "HOW OLD IS "G$" NOW?":INPUT A
110 PRINT "WHAT IS "P$" FAVORITE HOBBY":INPUT H$
120 PRINT "WHAT IS "P$" NICKNAME":INPUT NN$
130 PRINT "PRESS ANY KEY TO START THE JOKE."
140 GET S$:IF S$="" THEN140
150 HOME
160 PRINT "PRESS A KEY AND I WILL TALK TO YOU"
170 GET S$:IFS$=""THEN170
180 PRINT "HELLO, LET ME TRY TO GUESS YOUR NAME."
190 PRINT "I'M THINKING..."
200 GOSUB 400
210 PRINT "YOU LOOK LIKE SOME TYPE OF "N$:GOSUB400
220 PRINT "BUT I HOPE YOU WON'T MIND IF I CALL YOU"
230 PRINT NN$:GOSUB400
240 PRINT "THE TOUCH OF YOUR FINGERS":PRINT
250 PRINT "LEADS ME TO BELIEVE YOU ARE AT LEAST":PRINT
260 PRINT A" YEARS OLD":GOSUB400
270 PRINT "AND YOU SMELL LIKE A PERSON FROM":PRINT
280 PRINT C$:GOSUB400
290 PRINT "THE CENTRAL COMPUTER TELLS ME YOU LIKE:":PRINT
300 PRINT H$:GOSUB400
310 PRINT "NOW, IT'S YOUR TURN TO ASK ME A QUESTION"
320 INPUT "TYPE YOUR QUESTION, THEN PRESS RETURN ";Q$
330 GOSUB400
340 PRINT "SORRY, "NN$" THAT'S TOO PERSONAL!"
350 PRINT "YOUR ACCESS TO THE MASTER COMPUTER HAS"
360 PRINT "BEEN TERMINATED UNTIL 1999":END
400 FOR T = 1 TO 4000:NEXTT:HOME:RETURN

```

continued on next page....

## APPLE

## PRACTICAL JOKE con't...

In the PRACTICAL JOKE program you are introduced to some new commands. Two are called GOSUB and RETURN (not the key). GOSUB 400 means, "goto the subroutine at 400". A SUBROUTINE is like a program, within a program. RETURN means "return to the main program". A SUBROUTINE always starts with GOSUB and ends with RETURN. This SUBROUTINE, line 400, causes a time delay and clears the screen. In lines 140 and 170 is another, new command, called GET. GET S\$ tells the computer to wait for a key to be pressed. In line 140, if no key is pressed, the computer waits at line 140. When a key is pressed, the program proceeds on to line 150.

## VARIABLES are:

T=time delay

N\$=victim's name

MF\$=male or female

G\$=he or she

C\$=city

A=age

H\$=hobby

NN\$=nickname

P\$=his or her

Q\$=question

S\$=get variable

## APPLE

## Bonus Program #10 MATH SHARPENER

This program is called the MATH SHARPENER. I've been saving this program for the whiz kids. You must be one, or you wouldn't be reading this. The MATH SHARPENER has been designed to quiz both the beginner, and the advanced, on basic math skills. Don't use a scratch pad and I assure you, you'll get a work out. You will receive instructions when you RUN the program.

```

10 REM***MATH SHARPENER***
20 S=0:HOME
30 PRINT"THIS IS A PROGRAM TO SHARPEN YOUR MATH"
40 PRINT
50 PRINT "PICK#: 1)EASY 2)HARD, THEN RETURN"
60 INPUT L
70 PRINT"THE SYMBOLS ARE:"
80 PRINT"+ ADD          - SUBTRACT"
90 PRINT"* MULTIPLY    / DIVIDE":PRINT
100 PRINT"EXAMPLES:"
110 PRINT"2+3=5          8-4=4"
120 PRINT"2*4=8          9/3=3":PRINT
130 PRINT"PICK THE NUMBER OF THE FUNCTION"
140 PRINT"YOU WANT TO PRACTICE:":PRINT
150 PRINT"1)ADD"
160 PRINT"2)SUBTRACT"
170 PRINT"3)MULTIPLY"
180 PRINT"4)DIVIDE"
190 INPUT D
200 HOME:PRINT"TYPE THE NUMBER OF QUESTIONS"
210 INPUT"YOU WANT, THEN PRESS RETURN ";Q
220 FOR T=1 TO Q
230 PRINT"ANSWER THE PROBLEM, THEN PRESS RETURN":PRINT
240 IF L=1 THEN A=INT(RND(1)*10)+1:B=INT(RND(1)*10)+1
250 IF L=2 THEN A=INT(RND(1)*200)+1:B=INT(RND(1)*200)+1
260 IF D=2 AND A<B THEN 240
270 IF D=4 AND A<B THEN 240
280 IF D=4 AND A/B <> INT(A/B) THEN 240
290 IF A=B OR B=1 THEN 240
300 IF D=1 THEN PRINT A+"B"="";:INPUTC
310 IF D=2 THEN PRINT A+"B"="";:INPUTC
320 IF D=3 AND L=1 THEN PRINT A*"B"="";:INPUT C
330 IF D=3 AND L=2 THEN PRINT INT(A/5)*"INT(B/5)"="";:INPUTC
340 IF D=4 THEN PRINT A/"B"="";:INPUTC

```

continued on next page...

## APPLE

```

350 PRINT
360 IF D=1 AND C=A+B THEN S=S+1:PRINT "ALRIGHT!":GOTO490
370 IF D=2 AND C=A-B THEN S=S+1:PRINT"RIGHT ON!":GOTO490
380 IF D=3 AND L=1 AND C=A*B THEN S=S+1:Z=1
390 IF Z=1 THEN PRINT "CORRECT!":Z=0:GOTO490
400 IF D=3 AND L=2 AND C=INT(A/5)*INT(B/5) THEN S=S+1:Z=2
410 IF Z=2 THEN PRINT "GREAT!":Z=0:GOTO490
420 IF D=4 AND C=A/B THEN S=S+1:PRINT"YOWSAH!":GOTO490
430 PRINT"YOU GOOFED. THE ANSWER WAS:";
440 IF D=1 THEN PRINT A+B
450 IF D=2 THEN PRINT A-B
460 IF D=3 AND L=1 THEN PRINT A*B
470 IF D=3 AND L=2 THEN PRINT INT(A/5)*INT(B/5)
480 IF D=4 THEN PRINT A/B
490 FOR W=1TO800:NEXT W:HOME:NEXT T
500 PRINT"YOUR SCORE IS:"S" OUT OF "Q" RIGHT!"
510 PRINT"PRESS ANY KEY TO CONTINUE"
520 GET A$: IF A$= ""THEN520
530 GOTO 10

```

The VARIABLES are:

L=easy/hard

D=function to practice

Q=# of questions

T=question loop

A=random number 1

B=random number 2

C=choice (your answer)

W=delay

S=score

A\$=get variable

Z=correct multiplication flag

If you have problems getting this program to work properly, double check lines 240-500. These are the lines where most of the mathematical processing takes place. In this section, if D=1, then you are adding; if D=2, you are subtracting; D=3, means multiplying; and D=4 signifies division. If L=1, then the quizzes are easy. If L=2, then the quizzes are hard. Should you have the desire to make the program more difficult, you can increase the RANDOM NUMBERS, in lines 240-250.

Atari 400/800

# Bonus Program #1 CARTOON ROBOT

Riddle: What's sweet, but square; high tech, yet down to earth; and brilliant, with the I.Q. of a doughnut?

Give up?

The answer is: The Fabulous KISSING ROBOT.

You'll key in a program using FOR and NEXT to make a cartoon. The command FOR and NEXT are used for counting.

```

14 PRINT CHR$(125)
16 PRINT "()******()"
18 PRINT "**          **"
20 PRINT "** (0)      (0) **"
22 PRINT "**          **"
24 PRINT "**      V      **"
26 PRINT "**          **"
28 PRINT "**      ---     **"
30 FOR T=1 TO 50:NEXT T
32 PRINT CHR$(125)
34 PRINT "()******()"
36 PRINT "**          **"
38 PRINT "** (0)      (-) **"
40 PRINT "**          **"
42 PRINT "**      V      **"
44 PRINT "**          **"
46 PRINT "**      O      **"
48 FOR T = 1 TO 50: NEXT T
50 PRINT CHR$(125)
52 PRINT "()******()"
54 PRINT "**          **"
56 PRINT "** (0)      (0) **"
58 PRINT "**          **"
60 PRINT "**      V      **"
62 PRINT "**          **"
64 PRINT "**      O      **"
66 FOR T = 1 TO 50: NEXT T
68 GOTO 14

```

This program works like a real cartoon. The robot is printed on the screen, and erased, three times. Each time it is printed, there are small changes made, which give the illusion of movement. Line 68 GOTO 14 starts the entire process over again. The FOR/NEXT commands are used as time delays between pictures. You can change the speed of the cartoon by changing the "50", in the FOR/NEXT lines, to a different number. Decreasing the number will make the cartoon faster.

## ATARI

## Bonus Program #2 FAMILY DECISION MAKER

How would you like to use your home computer for solving problems like, "Who will use the computer first, Jimmy or Bobby?"...or how about, "Should we use the t.v. to watch a movie, or play with the computer?". The FAMILY DECISION MAKER can help you solve these problems, and more. It will make the decision for you, by picking a random choice. All you have to do is to type in the options.

```

10 DIM O1$(50),O2$(50):PRINT CHR$(125)
20 PRINT "FAMILY DECISION MAKER"
30 FOR T = 1 TO 1500:NEXT T
40 PRINT CHR$(125)
50 PRINT "TYPE IN THE OPTIONS"
60 PRINT "AND THE COMPUTER WILL DECIDE"
70 PRINT "WHAT IS OPTION #1";:INPUT O1$
80 PRINT "WHAT IS OPTION #2";:INPUT O2$
90 PRINT "I'M THINKING IT OVER....."
100 FOR T = 1 TO 3000:NEXT T
110 C=INT(RND(1)*2)+1
120 PRINT CHR$(125)
130 PRINT "MY CHOICE IS:"
140 IF C = 1 THEN PRINT O1$
150 IF C = 2 THEN PRINT O2$

```

The computer makes its choice in line 110. Lines 90 and 100 are where the computer is, "thinking it over". You probably noticed that the computer isn't really "thinking it over". It's actually counting up to 3000, then executing line 110. The choice is printed on the screen in lines 130-150. DIM O1\$(50),O2\$(50) sets a maximum of fifty letters, in each choice.

The VARIABLES are:

T=time delay  
 C=choice  
 O1\$=option #1  
 O2\$=option #2

ATARI

## Bonus Program #3 REACTION TIMER

Here is a program to test your reaction time. When the computer says "GO!", you must press the BREAK key as quickly as you can. Compare your score with the chart in the program. Good luck!

```

10 DIM A$(1):PRINT CHR$(125)
20 PRINT "TEST YOUR REACTION TIME"
30 PRINT "AGAINST THE COMPUTER."
40 PRINT "WHEN THE COMPUTER SAYS 'GO!'"
50 PRINT "PRESS THE BREAK KEY"
60 PRINT "YOUR SCORE IS THE HIGHEST NUMBER YOU SEE"
70 PRINT:PRINT "01-10=LIGHTNIN' 10-20=QUICK!"
80 PRINT "20-30=AVERAGE 30-50=NAPPING"
90 PRINT:PRINT:PRINT "PRESS RETURN"
100 PRINT "WHEN YOU ARE READY"
110 INPUT A$
120 PRINT CHR$(125):PRINT "ON YOUR MARK"
130 FOR T = 1 TO 1000:NEXT T:PRINT "GET SET!"
140 FOR T = 1 TO INT(RND(1)*5000):NEXT T
150 PRINT CHR$(125):PRINT "GO!"
160 FOR T = 1 TO 50:PRINT T:NEXT T
170 PRINT "SOMEONE WAKE THIS PERSON UP!"

```

The VARIABLE, of the FOR/NEXT statement in line 140, equals a RANDOM INTEGER between one and five thousand. This causes the time delay to be different each time the program is RUN. When you press the BREAK key the computer will say, "STOPPED AT 160". This is normal for the program. Your score is the highest number you see. Type RUN and press RETURN to play again.

## ATARI

## Bonus Program #4 M.P.G. RECORDER

If you're like me, you never take the trouble to figure out your car's miles per gallon (M.P.G.). Even having a calculator handy has never helped, though there are only three basic numbers to calculate. This is one more instance in which wanting to use my computer motivates me to do the fairly simple task I've managed to ignore. My mechanic tells me that I should check my M.P.G. after every five fill-ups. That way, if my M.P.G. starts dropping, I can take my car in for a checkup...before it's too late.

```

10 PRINT CHR$(125)
20 PRINT "MPG CALCULATOR"
30 FOR T = 1 TO 1500:NEXT T
40 PRINT CHR$(125)
50 PRINT "THIS IS A PROGRAM TO FIGURE OUT"
60 PRINT "THE MILES PER GALLON YOUR CAR GETS"
70 PRINT "HOW MANY MILES HAVE YOU DRIVEN"
80 PRINT "DURING THE PAST FIVE FILL-UPS"
90 INPUT M
100 PRINT "HOW MANY GALLONS OF GAS DID YOU USE"
110 PRINT "IN THE PAST FIVE FILL-UPS"
120 INPUT G
130 MPG=M/G
140 PRINT "YOU HAVE BEEN GETTING ";MPG
150 PRINT "MILES PER GALLON"

```

Notice that we used G as the VARIABLE for gas, M as the VARIABLE for miles, and MPG as the VARIABLE for miles per gallon. In line 130,  $MPG=M/G$  means miles per gallon equals miles divided by gallons.

## ATARI

## Bonus Program #5 COUPON CALCULATOR

Computers are pretty good at solving problems and presenting the results in a manner which is easy to read. This program can be used to display the amount of money you will save with your shopping coupons.

```

10 PRINT CHR$(125)
20 PRINT "COUPON CALCULATOR"
30 PRINT:PRINT "TO FIND OUT HOW MUCH YOU'LL SAVE"
40 PRINT "WITH YOUR SHOPPING COUPONS"
50 PRINT "ANSWER THE FOLLOWING QUESTIONS"
60 PRINT "THEN PRESS THE RETURN KEY"
70 PRINT:PRINT "HOW MANY COUPONS DO YOU HAVE"
80 INPUT C
90 PRINT:PRINT "ENTER THE AMOUNT OF A COUPON"
100 PRINT "DON'T USE A DOLLAR SIGN"
110 PRINT "DO USE A DECIMAL POINT":PRINT
120 FOR E = 1 TO C: PRINT "ENTER VALUE OF COUPON #";E
130 INPUT A
140 T=T+A
150 NEXT E
160 PRINT:PRINT "$";T;" WILL BE SAVED"

```

The VARIABLES in this program are:

C=number of coupons

E=coupon#

A=value of coupons

T=total value of coupons

Line 140 adds up the total, each time a value is entered into the computer. Line 160 prints the total value to be saved.

ATARI

## Bonus Program #6 SPORTS FORCASTER

The SPORTS FORECASTER can be a handy program if you enjoy sports. This program will take a team's current record and project, based on winning percentage, what the team's record will be at the end of the season.

```

10 PRINT CHR$(125)
20 PRINT "SPORTS FORECASTER"
30 FOR Z = 1 TO 1500:NEXT Z:PRINT
40 PRINT "THIS PROGRAM WILL FORECAST A TEAM'S"
50 PRINT "FINAL WIN AND LOSS RECORD"
60 PRINT "BASED ON ITS CURRENT RECORD"
70 PRINT:PRINT "ANSWER EACH QUESTION"
80 PRINT "THEN PRESS RETURN"
90 PRINT:PRINT "HOW MANY GAMES DOES THE TEAM PLAY"
100 INPUT T
110 PRINT:PRINT"HOW MANY WINS DO THEY HAVE NOW"
120 INPUT W
130 PRINT "HOW MANY LOSSES DO THEY HAVE NOW"
140 INPUT L
150 P=W/(W+L):Y=T*P:D=T-Y
160 PRINT:PRINT "END OF THE SEASON PROJECTION:"
170 PRINT:PRINT "WINS=";INT(Y);" LOSSES=";INT(D)+1

```

The VARIABLES are:

Z=time delay variable

T=total games in season

W=games won

L=games lost

P=percentage of games won

Y=end of year games projected won

D=end of year games projected lost

The forecast is completed in line 150 when the winning percentage (P) is established by dividing the number of completed games (W+L) into the total games won so far (W). The total wins for the year is estimated by multiplying the amount of games in the season (T) by the winning percentage (P). The year end losses are determined by subtracting the end of year projected games won (Y) from the total games in the season (T).

## ATARI

## Bonus Program #7 ELECTION RETURNS

Stage a mock primary, in which there are four candidates. Twenty five precincts report, one at a time. Running totals are printed as each reports. When all the returns are in, the computer displays the winner. This is a great way to get a feel for computerized election returns.

```

10 DIM N1$(15),N2$(15),N3$(15),N4$(15)
20 PRINT CHR$(125)
30 PRINT "ELECTION RETURNS"
40 FOR T=1 TO 1500:NEXT T
50 PRINT "INPUT THE NAME OF CANDIDATE #1";:INPUT N1$
60 PRINT "INPUT THE NAME OF CANDIDATE #2";:INPUT N2$
70 PRINT "INPUT THE NAME OF CANDIDATE #3";:INPUT N3$
80 PRINT "INPUT THE NAME OF CANDIDATE #4";:INPUT N4$
90 PRINT CHR$(125)
100 PRINT "THE POLLS JUST CLOSED"
110 PRINT "AND THE RESULTS ARE COMING IN"
120 FOR P=1 TO 25
130 N1=INT(RND(1)*999):N2=INT(RND(1)*999)
140 N3=INT(RND(1)*999):N4=INT(RND(1)*999)
150 FOR T=1 TO 2000:NEXT T
160 PRINT CHR$(125)
170 PRINT "PRECINCT#";P;" RESULTS:"
180 PRINT N1;"-";N1$:T1=N1+T1
190 PRINT N2;"-";N2$:T2=N2+T2
200 PRINT N3;"-";N3$:T3=N3+T3
210 PRINT N4;"-";N4$:T4=N4+T4
220 FOR T=1 TO 1500:NEXT T
230 PRINT
240 PRINT "CURRENT TOTALS:"
250 PRINT T1;"-";N1$
260 PRINT T2;"-";N2$
270 PRINT T3;"-";N3$
280 PRINT T4;"-";N4$
290 FOR T=1 TO 2000:NEXT T
300 NEXT P
310 PRINT "ALL RETURNS IN, AND"
320 IF T1>T2 AND T1>T3 AND T1>T4 THEN PRINT N1$
330 IF T2>T1 AND T2>T3 AND T2>T4 THEN PRINT N2$
340 IF T3>T1 AND T3>T2 AND T3>T4 THEN PRINT N3$
350 IF T4>T1 AND T4>T2 AND T4>T3 THEN PRINT N4$
360 PRINT "IS THE WINNER!!!"

```

continued on next page...

## ATARI

Bonus Program #7 continued...

The VARIABLES are:

N1\$=candidate #1 name  
N2\$=candidate #2 name  
N3\$=candidate #3 name  
N4\$=candidate #4 name  
N1=can. #1 local votes  
N2=can. #2 local votes  
N3=can. #3 local votes  
N4=can. #4 local votes  
T1=can. #1 total votes  
T2=can. #2 total votes  
T3=can. #3 total votes  
T4=can. #4 total votes  
T=time delay  
P=precinct loop

The DIM statements, in line 10, set the maximum number of letters in each candidate's name to fifteen. Changing line 120 will allow you to have more precincts. The 999's, in lines 130 and 140, can be modified to produce a higher number of votes in each precinct. If the program runs too slow for you, just change the time delays in lines 150, 220, and 290. Try 1000, or maybe even 500, if you are a quick reader.

## ATARI

## Bonus Program #8 NUMBER GAMES FOR TWO

Computers are great for playing games. They can be programmed to make games, both unpredictable and exciting. Here is a super game for two people. The computer "pulls a number out of its hat", and the players take turns trying to guess the number. The player with the most correct guesses, after seven rounds, is the champ. Switch sides after seven rounds. You will be surprised at the strategies involved.

```

10 DIM P1$(15),P2$(15):PRINT CHR$(125)
20 PRINT "THIS IS A NUMBER GAME FOR TWO PEOPLE"
30 PRINT "THE COMPUTER PICKS A NUMBER BETWEEN 1 AND 500.."
40 PRINT "THE PLAYERS TAKE TURNS GUESSING THE NUMBER"
50 PRINT "UNTIL SOMEONE GUESSES THE NUMBER"
60 PRINT "THE PLAYER GUESSING THE MOST NUMBERS,"
70 PRINT "AFTER 7 ROUNDS, IS THE WINNER"
80 PRINT "WHAT IS PLAYER #1'S NAME";:INPUT P1$
90 PRINT "WHAT IS PLAYER #2'S NAME";:INPUT P2$
100 R=R+1:N = INT (RND(1) * 500) + 1
110 IF R>1 THEN PRINT"THE SCORE IS ";P1$;"=";P1;"
";P2$;"=";P2
120 FOR T = 1 TO 2500:NEXT T
130 IF R > 7 THEN 260
140 FOR T = 1 TO 1000: NEXT T
150 PRINT CHR$(125) : PRINT "ROUND ";R;" , ";P1$;" 'S TURN"
160 PRINT "WHAT IS YOUR GUESS ";:INPUT G1
170 IF G1 < N THEN PRINT "TOO LOW ";P1$:GOTO 200
180 IF G1 > N THEN PRINT "TOO HIGH":GO TO 200
190 PRINT "YOU GOT IT ";P1$:P1 = P1 + 1:GOTO 100
200 FOR T = 1 TO 1000: NEXT T
210 PRINT CHR$(125):PRINT "ROUND ";R;" , ";P2$;" 'S TURN"
220 PRINT:PRINT "WHAT IS YOUR GUESS";:INPUT G2
230 IF G2 < N THEN PRINT "TOO LOW":GOTO 140
240 IF G2 > N THEN PRINT "TOO HIGH":GOTO 140
250 PRINT "YOU GOT IT ";P2$:P2 = P2 + 1:GOTO 100
260 PRINT CHR$(125):FOR T = 1 TO 1000: NEXT T
270 IF P1 > P2 THEN PRINT P1$;" CREAMED ";P2$;" ";P1;" TO
";P2:END
280 PRINT P2$;" WASTED ";P1$;" ";P2;" TO ";P1

```

The VARIABLES are:

P1\$=player #1

P2\$=player #2

P1=player #1 score

P2=player #2 score

G1=player #1 guess

G2=player #2 guess

R=round#

T=time delay variable

N=secret number

The secret number (N) is picked in line 100. To alter the limits of the secret number, you can change the 500 to a larger or smaller number. Try 10000, for instance. In several places you may notice symbols like this: > or this: < . The symbol > means "greater than" and < means "less than". Can you make this game work with four players?

## ATARI

## Bonus Program #9 PRACTICAL JOKER

Are you ready for some laughs? If so, Bonus Program #9 is the one for you. It's called the PRACTICAL JOKE PROGRAM. Here's how it works:

You type the program into your computer while the victim isn't around. RUN the program. The computer will ask you questions about the victim. You answer all the questions until the computer says, "PRESS RETURN TO START THE JOKE". Press a key and the joke is ready for the victim. When the victim comes back, the computer will seem to know all about him/her. Just say that you're hooked up to the MASTER COMPUTER, and it knows EVERYTHING!!!

```

10 DIM N$(25),MF$(10),C$(25),H$(25),NN$(25),S$(
1),Q$(25),G$(10),P$(10)
20 PRINT CHR$(125)
30 PRINT "PRACTICAL JOKE PROGRAM"
40 FOR T=1 TO 1000:NEXT T:PRINT CHR$(125)
50 PRINT "WHAT IS THE VICTIM'S NAME";:INPUT N$
60 PRINT "IS THE VICTIM MALE OR FEMALE";:INPUT MF$
70 IF MF$= "MALE" OR MF$= "M" THEN G$= "HE":P$= "HIS"
80 IF MF$= "FEMALE" OR MF$= "F" THEN G$= "SHE":P$= "HER"
90 PRINT "WHAT CITY IS ";G$;" FROM?":INPUT C$
100 PRINT "HOW OLD IS ";G$;" NOW?":INPUT A
110 PRINT "WHAT IS ";P$;" FAVORITE HOBBY":INPUT H$
120 PRINT "WHAT IS ";P$;" NICKNAME":INPUT NN$
130 PRINT "PRESS RETURN TO START THE JOKE."
140 INPUT S$
150 PRINT CHR$(125)
160 PRINT "PRESS RETURN AND I WILL TALK TO YOU"
170 INPUT S$
180 PRINT "HELLO, LET ME TRY TO GUESS YOUR NAME."
190 PRINT "I'M THINKING..."
200 GOSUB 400
210 PRINT "YOU LOOK LIKE SOME TYPE OF ";N$:GOSUB400
220 PRINT "BUT I HOPE YOU WON'T MIND IF I CALL YOU"
230 PRINT NN$:GOSUB400
240 PRINT "THE TOUCH OF YOUR FINGERS":PRINT
250 PRINT "LEADS ME TO BELIEVE YOU ARE AT LEAST":PRINT
260 PRINT A;" YEARS OLD":GOSUB400
270 PRINT "AND YOU SMELL LIKE A PERSON FROM":PRINT
280 PRINT C$:GOSUB400
290 PRINT "THE CENTRAL COMPUTER TELLS ME YOU LIKE:":PRINT
300 PRINT H$:GOSUB400
310 PRINT "NOW, IT'S YOUR TURN TO ASK ME A QUESTION"
320 PRINT "TYPE YOUR QUESTION, THEN PRESS RETURN";:INPUT Q$
330 GOSUB400
340 PRINT "SORRY, ";NN$;" THAT'S TOO PERSONAL!"
350 PRINT "YOUR ACCESS TO THE MASTER COMPUTER HAS"
360 PRINT "BEEN TERMINATED UNTIL 1999":END
400 FOR T = 1 TO 4000:NEXTT:PRINT CHR$(125):RETURN

```

continued on next page....

## ATARI

In the PRACTICAL JOKE program you are introduced to some new commands. Two are called GOSUB and RETURN (not the key). GOSUB 400 means, "goto the subroutine at 400". A SUBROUTINE is like a program, within a program. RETURN means "return to the main program". A SUBROUTINE always starts with GOSUB and ends with RETURN. This SUBROUTINE, line 400, causes a time delay and clears the screen.

## VARIABLES are:

T=time delay

N#=victim's name

MF#=male or female

G#=he or she

C#=city

A=age

H#=hobby

NN#=nickname

P#=his or her

Q#=question

## ATARI

## Bonus Program #10 MATH SHARPENER

This program is called the MATH SHARPENER. I've been saving this program for the whiz kids. You must be one, or you wouldn't be reading this. The MATH SHARPENER has been designed to quiz both the beginner, and the advanced, on basic math skills. Don't use a scratch pad and I assure you, you'll get a work out. You will receive instructions when you RUN the program.

```

10 DIM A$(1)
20 PRINT CHR$(125)
30 PRINT "THIS IS A PROGRAM TO SHARPEN YOUR MATH"
40 PRINT
50 PRINT "CHOOSE: 1)EASY 2)HARD, THEN PRESS RETURN"
60 INPUT L
70 PRINT "THE SYMBOLS ARE:"
80 PRINT "+ ADD          - SUBTRACT"
90 PRINT "* MULTIPLY     / DIVIDE":PRINT
100 PRINT "EXAMPLES:"
110 PRINT "2+3=5          8-4=4"
120 PRINT "4*2=8          9/3=3"
130 PRINT "PICK THE NUMBER OF THE FUNCTION"
140 PRINT "YOU WANT TO PRACTICE:":PRINT
150 PRINT "1)ADD"
160 PRINT "2)SUBTRACT"
170 PRINT "3)MULTIPLY"
180 PRINT "4)DIVIDE"
190 INPUT D
200 PRINT CHR$(125):PRINT "TYPE THE NUMBER OF QUESTIONS"
210 PRINT "YOU WANT, THEN PRESS RETURN":INPUT Q
220 FOR T=1 TO Q
230 PRINT "ANSWER THE PROBLEM, THEN PRESS RETURN":PRINT
240 IF L=1 THEN A=INT(RND(1)*10)+1:B=INT(RND(1)*10)+1
250 IF L=2 THEN A=INT(RND(1)*200)+1:B=INT(RND(1)*200)+1
260 IF D=2 AND A<B THEN 240
270 IF D=4 AND A<B THEN 240
280 IF D=4 AND A/B <> INT(A/B) THEN 240
290 IF A=B OR B=1 THEN 240
300 IF D=1 THEN PRINT A;"+";B;"=":INPUT C
310 IF D=2 THEN PRINT A;"-";B;"=":INPUT C
320 IF D=3 AND L=1 THEN PRINT A;"*";B;"=":INPUT C
330 IF D=3 AND L=2 THEN PRINT
    INT(A/5);"*";INT(B/5);"=":INPUT C
340 IF D=4 THEN PRINT A;"/";B;"=":INPUT C

```

continued on next page...

## ATARI

```

350 PRINT
360 IF D=1 AND C=A+B THEN S=S+1:PRINT "ALRIGHT!":GOTO490
370 IF D=2 AND C=A-B THEN S=S+1:PRINT"RIGHT ON!":GOTO490
380 IF D=3 AND L=1 AND C=A*B THEN S=S+1:Z=1
390 IF Z=1 THEN PRINT "CORRECT!":Z=0:GOTO490
400 IF D=3 AND L=2 AND C=INT(A/5)*INT(B/5) THEN S=S+1:Z=2
410 IF Z=2 THEN PRINT "GREAT!":Z=0:GOTO500
420 IF D=4 AND C=A/B THEN S=S+1:PRINT"YOWSAH!":GOTO490
430 PRINT"YOU GOOFED. THE ANSWER WAS:";
440 IF D=1 THEN PRINT A+B
450 IF D=2 THEN PRINT A-B
460 IF D=3 AND L=1 THEN PRINT A*B
470 IF D=3 AND L=2 THEN PRINT INT(A/5)*INT(B/5)
480 IF D=4 THEN PRINT A/B
490 FOR W=1TO800:NEXT W:PRINT CHR$(125):NEXT T
500 PRINT"YOUR SCORE IS:";S;" OUT OF ";Q;" RIGHT!"
510 PRINT"PRESS RETURN TO CONTINUE"
520 INPUT A$
530 S=0:GOTO 20

```

The VARIABLES are:

L=easy/hard

D=function to practice

Q=# of questions

T=question loop

A=random number 1

B=random number 2

C=choice (your answer)

W=delay

S=score

A\$=continue variable

Z=correct multiplication flag

If you have problems getting this program to work properly, double check lines 240-500. These are the lines where most of the mathematical processing takes place. In this section, if D=1, then you are adding; if D=2, you are subtracting; D=3, means multiplying; and D=4 signifies division. If L=1, then the quizzes are easy. If L=2, then the quizzes are hard. Should you have the desire to make the program more difficult, you can increase the RANDOM NUMBERS, in lines 240-250.

## COMMODORE 64 AND VIC-20

## Bonus Program #1 CARTOON ROBOT

Riddle: What's seet, but square; high tech, yet down to earth; and brilliant, with the I.Q. of a doughnut?

Give up?

The answer is: The fabulous KISSING ROBOT.

You'll key in a program using FOR and NEXT to make a cartoon. The commands FOR and NEXT are used for counting.

```

14 PRINT CHR$(147)
16 PRINT "()******()"
18 PRINT "**          **"
20 PRINT "** (O)      (O) **"
22 PRINT "**          **"
24 PRINT "**      V      **"
26 PRINT "**          **"
28 PRINT "**      ---     **"
30 FOR T = 1 TO 75: NEXT T
32 PRINT CHR$(147)
34 PRINT "()******()"
36 PRINT "**          **"
38 PRINT "** (O)      (-) **"
40 PRINT "**          **"
42 PRINT "**      V      **"
44 PRINT "**          **"
46 PRINT "**      O      **"
48 FOR T = 1 TO 75: NEXT T
50 PRINT CHR$(147)
52 PRINT "()******()"
54 PRINT "**          **"
56 PRINT "** (O)      (O) **"
58 PRINT "**          **"
60 PRINT "**      V      **"
62 PRINT "**          **"
64 PRINT "**      O      **"
66 FOR T = 1 TO 75: NEXT T
68 GOTO 14

```

This program works like a real cartoon. The robot is printed on the screen, and erased, three times. Each time it is printed, there are small changes made, which give the illusion of movement. Line 68 GOTO 14 starts the entire process over again. The FOR/NEXT commands are used as time delays between pictures. You can change the speed of the cartoon by changing the 75, in the FOR/NEXT lines, to a different number. Decreasing the number will make the cartoon faster.

## COMMODORE

## Bonus Program #2 FAMILY DECISION MAKER

How would you like to use your home computer for solving problems like, "Who will use the computer first, Jimmy or Bobby?"...or how about, "Should we use the t.v. to watch a movie, or play with the computer?". The FAMILY DECISION MAKER can help you solve these problems, and more. It will make the decision for you, by picking a random choice. All you have to do is to type in the options.

```

10 PRINT CHR$(147)
20 PRINT "FAMILY DECISION MAKER"
30 FOR T = 1 TO 1500:NEXT T
40 PRINT CHR$(147)
50 PRINT "TYPE IN THE OPTIONS"
60 PRINT "AND THE COMPUTER WILL DECIDE"
70 INPUT "WHAT IS OPTION #1";O1$
80 INPUT "WHAT IS OPTION #2";O2$
90 PRINT "I'M THINKING IT OVER....."
100 FOR T = 1 TO 3000:NEXT T
110 C=INT(RND(1)*2)+1
120 PRINT CHR$(147)
130 PRINT "MY CHOICE IS:"
140 IF C = 1 THEN PRINT O1$
150 IF C = 2 THEN PRINT O2$

```

In this program the computer makes its choice in line 110. Lines 90 and 100 are where the computer is, "thinking it over". You probably noticed that the computer isn't really "thinking it over". It's actually counting up to 3000, then executing line 110. The choice is printed on the screen in lines 130-150.

The VARIABLES are:

T=time delay

C=choice

O1\$=option #1

O2\$=option #2

## COMMODORE

## Bonus Program #3 REACTION TIMER

Here is a program to test your reaction time. When the computer says "GO!", you must press the RUN/STOP key as quickly as you can. Compare your score with the chart in the program. Good luck!

```

10 PRINT CHR$(147)
20 PRINT "TEST YOUR REACTION TIME"
30 PRINT "AGAINST THE COMPUTER."
40 PRINT "WHEN THE COMPUTER SAYS 'GO!'"
50 PRINT "PRESS THE RUN/STOP KEY"
60 PRINT "YOUR SCORE IS THE HIGHEST NUMBER YOU SEE"
70 PRINT:PRINT "01-10=LIGHTNIN' 10-20=QUICK!"
80 PRINT "20-30=AVERAGE 30-50=NAPPING"
90 PRINT:PRINT:PRINT "PRESS RETURN"
100 PRINT "WHEN YOU ARE READY"
110 INPUT A$
120 PRINT CHR$(147):PRINT "ON YOUR MARK"
130 FOR T = 1 TO 1000:NEXT T:PRINT "GET SET!"
140 FOR T = 1 TO INT(RND(1)*5000):NEXT T
150 PRINT CHR$(147):PRINT "GO!"
160 FOR T = 1 TO 50:PRINT T:NEXT T
170 PRINT "SOMEONE WAKE THIS PERSON UP!"

```

The VARIABLE, of the FOR/NEXT statement in line 140, equals a RANDOM INTEGER between one and five thousand. This causes the time delay to be different each time the program is RUN. When you press the RUN/STOP key the computer will say, "Break in 160". This is normal for the program. Your score is the highest number you see. Type RUN and press RETURN to play again.

## COMMODORE

## Bonus Program #4 M.P.G. RECORDER

If you're like me, you never take the trouble to figure out your car's miles per gallon (M.P.G.). Even having a calculator handy has never helped, though there are only three basic numbers to calculate. This is one more instance in which wanting to use my computer motivates me to do the fairly simple task I've managed to ignore. My mechanic tells me that I should check my M.P.G. after every five fill-ups. That way, if my M.P.G. starts dropping, I can take my car in for a checkup...before it's too late.

```

10 PRINT CHR$(147)
20 PRINT "MPG CALCULATOR"
30 FOR T = 1 TO 1500:NEXT T
40 PRINT CHR$(147)
50 PRINT "THIS IS A PROGRAM TO FIGURE OUT"
60 PRINT "THE MILES PER GALLON YOUR CAR GETS"
70 PRINT "HOW MANY MILES HAVE YOU DRIVEN"
80 PRINT "DURING THE PAST FIVE FILL-UPS"
90 INPUT M
100 PRINT "HOW MANY GALLONS OF GAS DID YOU USE"
110 PRINT "IN THE PAST FIVE FILL-UPS"
120 INPUT G
130 MPG=M/G
140 PRINT "YOU HAVE BEEN GETTING "MPG
150 PRINT "MILES PER GALLON"

```

Notice that we used G as the VARIABLE for gas, M as the VARIABLE for miles, and MPG as the VARIABLE for miles per gallon. In line 130,  $MPG=M/G$  means miles per gallon equals miles divided by gallons.

## COMMODORE

## Bonus Program #5 COUPON CALCULATOR

Computers are pretty good at solving problems and presenting the results in a manner which is easy to read. This program can be used to display the amount of money you will save with your shopping coupons.

```

10 PRINT CHR$(147)
20 PRINT "COUPON CALCULATOR"
30 PRINT:PRINT "TO FIND OUT HOW MUCH YOU'LL SAVE"
40 PRINT "WITH YOUR SHOPPING COUPONS"
50 PRINT "ANSWER THE FOLLOWING QUESTIONS"
60 PRINT "THEN PRESS THE RETURN KEY"
70 PRINT:PRINT "HOW MANY COUPONS DO YOU HAVE"
80 INPUT C
90 PRINT:PRINT "ENTER THE AMOUNT OF A COUPON"
100 PRINT "DON'T USE A DOLLAR SIGN"
110 PRINT "DO USE A DECIMAL POINT":PRINT
120 FOR E = 1 TO C: PRINT "ENTER VALUE OF COUPON #"E
130 INPUT A
140 T=T+A
150 NEXT E
160 PRINT:PRINT "$"T" WILL BE SAVED"

```

The VARIABLES in this program are:

C=number of coupons

E=coupon#

A=value of coupons

T=total value of coupons

Line 140 adds up the total, each time a value is entered into the computer. Line 160 prints the total value to be saved.

## COMMODORE

## Bonus Program #6 SPORTS FORECASTER

The SPORTS FORECASTER can be a handy program if you enjoy sports. This program will take a team's current record and project, based on winning percentage, what the team's record will be at the end of the season.

```

10 PRINT CHR$(147)
20 PRINT "SPORTS FORECASTER"
30 FOR Z = 1 TO 1500:NEXT Z:PRINT
40 PRINT "THIS PROGRAM WILL FORECAST A TEAM'S"
50 PRINT "FINAL WIN AND LOSS RECORD"
60 PRINT "BASED ON ITS CURRENT RECORD."
70 PRINT:PRINT "ANSWER EACH QUESTION"
80 PRINT "THEN PRESS RETURN"
90 PRINT:PRINT "HOW MANY GAMES DOES THE TEAM PLAY"
100 INPUT T
110 PRINT:PRINT"HOW MANY WINS DO THEY HAVE NOW"
120 INPUT W
130 PRINT "HOW MANY LOSSES DO THEY HAVE NOW"
140 INPUT L
150 P=W/(W+L):Y=T*P:D=T-Y
160 PRINT:PRINT "END OF THE SEASON PROJECTION:"
170 PRINT:PRINT "WINS="INT(Y)" LOSSES="INT(D)+1

```

The VARIABLES are:

Z=time delay variable

T=total games in season

W=games won

L=games lost

P=percentage of games won

Y=end of year games projected won

D=end of year games projected lost

The forecast is completed in line 150 when the winning percentage (P) is established by dividing the number of completed games (W+L) into the total games won so far (W). The total wins for the year is estimated by multiplying the amount of games in the season (T) by the winning percentage (P). The year end losses are determined by subtracting the end of year projected games won (Y) from the total games in the season (T).

## COMMODORE

## Bonus Program #7 SHOWER MONITOR

Getting into the shower, day after day, and finding cold water can be a drag. I'm sure that large families know what I'm talking about. Bonus Program #7 has been designed to whip morning bathroom confusion. It's called the SHOWER MONITOR. You type in the names and the computer picks the shower order.

```

10 PRINT CHR$(147)
20 PRINT "SHOWER MONITOR"
30 FOR T= 1 TO 2000:NEXT T:PRINT CHR$(147)
40 PRINT "THIS PROGRAM IS DESIGNED TO HELP"
50 PRINT "FAMILIES DECIDE, IN A FAIR WAY"
60 PRINT "THE ORDER IN WHICH THE SHOWER IS USED"
70 PRINT "IN THE MORNING."
80 PRINT:PRINT "EACH PERSON'S NAME IS TYPED INTO THE"
90 PRINT "COMPUTER. THEN THE COMPUTER RANDOMLY"
100 PRINT "CHOOSES THE ORDER (AS IF OUT OF A HAT). "
110 PRINT "TYPE EACH ANSWER, THEN PRESS RETURN."
120 PRINT:PRINT "HOW MANY PEOPLE IN YOUR FAMILY"
130 INPUT P
140 PRINT:PRINT "TYPE IN THE NAMES, ONE AT A TIME."
150 PRINT "THEN PRESS RETURN."
160 FOR H = 1 TO P
170 INPUT N$(H)
180 NEXT H
190 PRINT CHR$(147):PRINT "THIS IS THE SHOWER ORDER TODAY:"
200 PRINT:FOR R = 1 TO P
210 X=INT(RND(1)*P)+1
220 IF N$(X)="" THEN 210
230 PRINT N$(X)
240 N$(X)=""
250 NEXT R
260 GOTO 260

```

The VARIABLES are:

P=number of people in family

H=array parking lot#

X=random number

N\$(X)=name of person X in array

R=counting variable

You are probably wondering what an ARRAY is. An ARRAY is a computer parking lot. In an ARRAY you don't park cars. Rather, you park words and numbers. In this program we parked the name of each person in an ARRAY location (such as N\$(1)="Mom", N\$(2)="Larry", N\$(3)="Rick", etc). The names are loaded into the ARRAY in lines 160-180. The RANDOM shower order is determined in lines 200-250. Can you figure out why a name isn't picked more than once?

## COMMODORE

## Bonus Program #8 NUMBER GAME FOR TWO

Computers are great for playing games. They can be programmed to make games, both unpredictable and exciting. Here is a super game for two people. The computer "pulls a number out of its hat", and the players take turns trying to guess the number. The player with the most correct guesses, after seven rounds, is the champ. Switch sides after seven rounds. You will be surprised at the strategies involved.

```

10 PRINT CHR$(147)
20 PRINT "THIS IS A NUMBER GAME FOR TWO PEOPLE"
30 PRINT "THE COMPUTER PICKS A NUMBER BETWEEN 1 AND 500."
40 PRINT "THE PLAYERS TAKE TURNS GUESSING THE NUMBER"
50 PRINT "UNTIL SOMEONE GUESSES THE NUMBER"
60 PRINT "THE PLAYER GUESSING THE MOST NUMBERS,"
70 PRINT "AFTER 7 ROUNDS, IS THE WINNER"
80 PRINT : INPUT "WHAT IS PLAYER #1'S NAME";P1$
90 PRINT : INPUT "WHAT IS PLAYER #2'S NAME";P2$
100 R=R+1:N = INT (RND(1) * 500) +1
110 IF R>1 THEN PRINT"THE SCORE IS "P1$="P1" "P2$="P2
120 FOR T = 1 TO 2500:NEXT T
130 IF R > 7 THEN 260
140 FOR T = 1 TO 1000: NEXT T
150 PRINT CHR$(147) : PRINT "ROUND "R", "P1$"'S TURN"
160 PRINT : INPUT "WHAT IS YOUR GUESS";G1
170 IF G1 < N THEN PRINT "TOO LOW "P1$:GOTO 200
180 IF G1 > N THEN PRINT "TOO HIGH":GOTO 200
190 PRINT "YOU GOT IT "P1$:P1 = P1 + 1:GOTO 100
200 FOR T = 1 TO 1000: NEXT T
210 PRINT CHR$(147):PRINT "ROUND "R", "P2$"'S TURN"
220 PRINT:INPUT "WHAT IS YOUR GUESS";G2
230 IF G2 < N THEN PRINT "TOO LOW":GOTO 140
240 IF G2 > N THEN PRINT "TOO HIGH":GOTO 140
250 PRINT "YOU GOT IT "P2$:P2 = P2 +1:GOTO 100
260 PRINT CHR$(147):FOR T = 1 TO 1000: NEXT T
270 IF P1 > P2 THEN PRINT P1$" CREAMED "P2$" "P1" TO "P2:END
280 PRINT P2$" WASTED "P1$" "P2" TO "P1

```

The VARIABLES are:

P1\$=player #1

P2\$=player #2

P1=player #1 score

P2=player #2 score

G1=player #1 guess

G2=player #2 guess

R=round#

T=time delay variable

N=secret number

The secret number (N) is picked in line 100. To alter the limits of the secret number, you can change the 500 to a larger or smaller number. Try 10000, for instance. In several places you may notice symbols like this: > or this: < . The symbol > means "greater than" and < means "less than". Can you make this game work with four players?

## COMMODORE

## Bonus Program #9 PRACTICAL JOKER

Are you ready for some laughs? If so, Bonus Program #9 is the one for you. It's called the PRACTICAL JOKE PROGRAM. Here's how it works:

You type the program into your computer while the victim isn't around. RUN the program. The computer will ask you questions about the victim. You answer all the questions until the computer says, "PRESS ANY KEY TO START THE JOKE". Press a key and the joke is ready for the victim. When the victim comes back, the computer will seem to know all about him/her. Just say that you're hooked up to the MASTER COMPUTER, and it knows EVERYTHING!!!

```

10 PRINT CHR$(147)
20 PRINT "PRACTICAL JOKE PROGRAM"
30 FOR T=1 TO 1000:NEXT T
40 PRINT CHR$(147)
50 INPUT "WHAT IS THE VICTIM'S NAME";N$
60 INPUT "IS THE VICTIM MALE OR FEMALE";MF$
70 IF MF$= "MALE" OR MF$= "M" THEN G$= "HE":P$= "HIS"
80 IF MF$= "FEMALE" OR MF$= "F" THEN G$= "SHE":P$= "HER"
90 PRINT "WHAT CITY IS "G$" FROM?":INPUT C$
100 PRINT "HOW OLD IS "G$" NOW?":INPUT A
110 PRINT "WHAT IS "P$" FAVORITE HOBBY":INPUT H$
120 PRINT "WHAT IS "P$" NICKNAME":INPUT NN$
130 PRINT "PRESS ANY KEY TO START THE JOKE."
140 GET S$:IF S$="" THEN140
150 PRINT CHR$(147)
160 PRINT "PRESS A KEY AND I WILL TALK TO YOU"
170 GET S$:IFS$=""THEN170
180 PRINT "HELLO, LET ME TRY TO GUESS YOUR NAME."
190 PRINT "I'M THINKING..."
200 GOSUB 400
210 PRINT "YOU LOOK LIKE SOME TYPE OF "N$:GOSUB400
220 PRINT "BUT I HOPE YOU WON'T MIND IF I CALL YOU"
230 PRINT NN$:GOSUB400
240 PRINT "THE TOUCH OF YOUR FINGERS":PRINT
250 PRINT "LEADS ME TO BELIEVE YOU ARE AT LEAST":PRINT
260 PRINT A" YEARS OLD":GOSUB400
270 PRINT "AND YOU SMELL LIKE A PERSON FROM":PRINT
280 PRINT C$:GOSUB400
290 PRINT "THE CENTRAL COMPUTER TELLS ME YOU LIKE:":PRINT
300 PRINT H$:GOSUB400
310 PRINT "NOW, IT'S YOUR TURN TO ASK ME A QUESTION"
320 INPUT "TYPE YOUR QUESTION, THEN PRESS RETURN";Q$
330 GOSUB400
340 PRINT "SORRY, "NN$" THAT'S TOO PERSONAL!"
350 PRINT "YOUR ACCESS TO THE MASTER COMPUTER HAS"
360 PRINT "BEEN TERMINATED UNTIL 1999":END
400 FOR T = 1 TO 4000:NEXTT:PRINT CHR$(147):RETURN

```

continued on next page....

## COMMODORE

In the PRACTICAL JOKE program you are introduced to some new commands. Two are called GOSUB and RETURN (not the key). GOSUB 400 means, "goto the subroutine at 400". A SUBROUTINE is like a program, within a program. RETURN means "return to the main program". A SUBROUTINE always starts with GOSUB and ends with RETURN. This SUBROUTINE, line 400, causes a time delay and clears the screen. In lines 140 and 170 is another, new command, called GET. GET S\$ tells the computer to wait for a key to be pressed. In line 140, if no key is pressed, the computer waits at line 140. When a key is pressed, the program proceeds on to line 150.

## VARIABLES are:

T=time delay  
N\$=victim's name  
MF\$=male or female  
G\$=he or she  
C\$=city  
A=age  
H\$=hobby  
NN\$=nickname  
P\$=his or her  
Q\$=question  
S\$=get variable

## COMMODORE

## Bonus Program #10 MATH SHARPENER

This program is called the MATH SHARPENER. I've been saving this program for the whiz kids. You must be one, or you wouldn't be reading this. The MATH SHARPENER has been designed to quiz both the beginner, and the advanced, on basic math skills. Don't use a scratch pad and I assure you, you'll get a work out. You will receive instructions when you RUN the program.

```

10 REM***MATH SHARPENER***
20 PRINT CHR$(147)
30 PRINT"THIS IS A PROGRAM TO SHARPEN YOUR MATH"
40 PRINT
50 PRINT "CHOOSE:1)EASY 2)HARD, THEN PRESS RETURN"
60 INPUT L
70 PRINT"THE SYMBOLS ARE:"
80 PRINT"+ ADD          - SUBTRACT"
90 PRINT"* MULTIPLY    / DIVIDE":PRINT
100 PRINT"EXAMPLES:"
110 PRINT"2+3=5          8-4=4"
120 PRINT"2*4=8          9/3=3":PRINT
130 PRINT"PICK THE NUMBER OF THE FUNCTION"
140 PRINT"YOU WANT TO PRACTICE:":PRINT
150 PRINT"1)ADD"
160 PRINT"2)SUBTRACT"
170 PRINT"3)MULTIPLY"
180 PRINT"4)DIVIDE"
190 INPUT D
200 PRINT CHR$(147):PRINT"TYPE THE NUMBER OF QUESTIONS"
210 INPUT"YOU WANT, THEN PRESS RETURN";Q
220 FOR T=1 TO Q
230 PRINT"ANSWER THE PROBLEM, THEN PRESS RETURN":PRINT
240 IF L=1 THEN A=INT(RND(1)*10)+1:B=INT(RND(1)*10)+1
250 IF L=2 THEN A=INT(RND(1)*200)+1:B=INT (RND(1)*200)+1
260 IF D=2 AND A<B THEN 240
270 IF D=4 AND A<B THEN 240
280 IF D=4 AND A/B <> INT(A/B) THEN 240
290 IF A=B OR B=1 THEN 240
300 IF D=1 THEN PRINT A"+"B="";:INPUT C
310 IF D=2 THEN PRINT A "-"B="";:INPUT C
320 IF D=3 AND L=1 THEN PRINT A*"B="";:INPUT C
330 IF D=3 AND L=2 THEN PRINT INT(A/5)."*"INT(B/5)="";:INPUT C
340 IF D=4 THEN PRINT A"/"B="";:INPUT C

```

continued on next page...

## COMMODORE

```

350 PRINT
360 IF D=1 AND C=A+B THEN S=S+1:PRINT "ALRIGHT!":GOTO490
370 IF D=2 AND C=A-B THEN S=S+1:PRINT"RIGHT ON!":GOTO490
380 IF D=3 AND L=1 AND C=A*B THEN S=S+1:Z=1
390 IF Z=1 THEN PRINT "CORRECT!":Z=0:GOTO490
400 IF D=3 AND L=2 AND C=INT(A/5)*INT(B/5) THEN S=S+1:Z=2
410 IF Z=2 THEN PRINT "GREAT!":Z=0:GOTO500
420 IF D=4 AND C=A/B THEN S=S+1:PRINT"YOWSAH!":GOTO490
430 PRINT"YOU GOOFED. THE ANSWER WAS:";
440 IF D=1 THEN PRINT A+B
450 IF D=2 THEN PRINT A-B
460 IF D=3 AND L=1 THEN PRINT A*B
470 IF D=3 AND L=2 THEN PRINT INT(A/5)*INT(B/5)
480 IF D=4 THEN PRINT A/B
490 FOR W=1TO800:NEXT W:PRINT CHR$(147):NEXT T
500 PRINT"YOUR SCORE IS:"S" OUT OF "Q" "RIGHT!"
510 PRINT"PRESS ANY KEY TO CONTINUE"
520 GET A$: IF A$= ""THEN520
530 S=0:GOTO 10

```

The VARIABLES are:

L=easy/hard

D=function to practice

Q=# of questions

T=question loop

A=random number 1

B=random number 2

C=choice (your answer)

W=delay

S=score

A\$=get variable

Z=correct multiplication flag

If you have problems getting this program to work properly, double check lines 240-500. These are the lines where most of the mathematical processing takes place. In this section, if D=1, then you are adding; if D=2, you are subtracting; D=3, means multiplying; and D=4 signifies division. If L=1, then the quizzes are easy. If L=2, then the quizzes are hard. Should you have the desire to make the program more difficult, you can increase the RANDOM NUMBERS, in lines 240-250.

TI 99/4A

## Bonus Program #1 CARTOON ROBOT

BLAST OFF! is a computer cartoon in which the count down and blast off of a rocket takes place on your t.v. screen.

```

10 CALL CLEAR
20 FOR C=10 TO 1 STEP - 1
30 PRINT C
40 FOR T=1 TO 200
50 NEXT T
60 CALL CLEAR
70 NEXT C
80 CALL CLEAR
90 PRINT "BLAST OFF!!!!"
100 FOR T=1 TO 200
110 NEXT T

```

The above lines tell the computer to count down, from 10 to 1 then print, "BLASTOFF!".

Now add these to the lines above:

```

120 CALL CLEAR
130 PRINT "      ^"
140 PRINT "    /  \"
150 PRINT "   !  !"
160 PRINT "  !U!"
170 PRINT " !  !"
180 PRINT " !S!"
190 PRINT " !  !"
200 PRINT " !A!"
210 PRINT " !  !"
220 PRINT "  /  \"
230 PRINT " /  \"
240 PRINT:.....

```

To make the nose cone on the rocket, hold down the SHIFT key and press the NUMBER 6 key. The slanted lines, on the right side of the rocket, are created by holding down the FCTN key and pressing Z key. The slanted lines on the left side of the rocket can be printed by pressing the / key (just above the ENTER key). In line 240 you need to make at least twenty-five COLONS. Each COLON tells the computer to print a blank line. Type RUN, then press ENTER to see the rocket BLAST OFF!

TI 99/4A

## Bonus Program #2 FAMILY DECISION MAKER

How would you like to use your home computer for solving problems like, "Who will use the computer first, Jimmy or Bobby?"...or how about, "Should we use the t.v. to watch a movie, or play with the computer?". The FAMILY DECISION MAKER can help you solve these problems, and more. It will make the decision for you, by picking a random choice. All you have to do is to type in the options.

```

10 RANDOMIZE
20 CALL CLEAR
30 PRINT "FAMILY DECISION MAKER"
40 FOR T=1 TO 1500
50 NEXT T
60 CALL CLEAR
70 PRINT "TYPE IN THE OPTIONS AND"
80 PRINT "THE COMPUTER WILL DECIDE."
90 PRINT "WHAT IS OPTION #1"
100 INPUT O1$
110 PRINT "WHAT IS OPTION #2"
120 INPUT O2$
130 PRINT "I'M THINKING IT OVER....."
140 FOR T = 1 TO 1000
150 NEXT T
160 C=INT(RND*2)+1
170 CALL CLEAR
180 PRINT "MY CHOICE IS:"
190 IF C=2 THEN 220
200 PRINT O1$
210 END
220 PRINT O2$

```

In this program the computer makes its choice in line 160. Lines 130 to 150 are where the computer is, "thinking it over". You probably noticed that the computer isn't really "thinking it over". It's actually counting up to 1000, then executing line 160. The choice is printed on the screen in lines 180-220.

The VARIABLES are:

T=time delay  
 C=choice  
 O1\$=option #1  
 O2\$=option #2

TI 99/4A

## Bonus Program #3 M.P.G. RECORDER

If you're like me, you never take the trouble to figure out your car's miles per gallon (M.P.G.). Even having a calculator handy has never helped, though there are only three basic numbers to calculate. This is one more instance in which wanting to use my computer motivates me to do the fairly simple task I've managed to ignore. My mechanic tells me that I should check my M.P.G. after every five fill-ups. That way, if my M.P.G. starts dropping, I can take my car in for a checkup...before it's too late.

```
10 CALL CLEAR
20 PRINT "MPG CALCULATOR"
30 FOR T=1 TO 1500
40 NEXT T
50 CALL CLEAR
60 PRINT "THIS PROGRAM DETERMINES"
70 PRINT "THE MPG YOUR CAR GETS."
80 PRINT "HOW MANY MILES DID YOU DRIVE"
90 PRINT "DURING THE PAST 5 FILL-UPS"
100 INPUT M
110 PRINT "HOW MANY GALLONS OF GAS DID YOU"
120 PRINT "USE IN THE PAST 5 FILL-UPS"
130 INPUT G
140 MPG=M/G
150 PRINT "YOU HAVE BEEN GETTING";MPG
160 PRINT "MILES PER GALLON"
```

Notice that we used G as the VARIABLE for gas, M as the VARIABLE for miles, and MPG as the VARIABLE for miles per gallon. In line 140,  $MPG=M/G$  means miles per gallon equals miles divided by gallons.

TI 99/4A

## Bonus Program #4 COUPON CALCULATOR

Computers are pretty good at solving problems and presenting the results in a manner which is easy to read. This program can be used to display the amount of money you will save with your shopping coupons.

```

10 CALL CLEAR
20 PRINT "COUPON CALCULATOR"
30 PRINT
40 PRINT "FIND OUT WHAT YOU'LL SAVE"
50 PRINT "WITH YOUR SHOPPING COUPONS"
60 PRINT "ANSWER THE QUESTION"
70 PRINT "THEN PRESS THE ENTER KEY"
80 PRINT
90 PRINT "HOW MANY COUPONS DO YOU HAVE"
100 INPUT C
110 PRINT
120 PRINT "TO ENTER VALUE OF A COUPON"
130 PRINT "DON'T USE A DOLLAR SIGN"
140 PRINT "DO USE A DECIMAL POINT"
150 PRINT
160 FOR E = 1 TO C
170 PRINT "ENTER VALUE OF COUPON #";E
180 INPUT A
190 T=T+A
200 NEXT E
210 PRINT
220 PRINT "$";T;"WILL BE SAVED"

```

The VARIABLES in this program are:

C=number of coupons

E=coupon#

A=value of coupons

T=total value of coupons

Line 190 adds up the total, each time a value is entered into the computer. Line 220 prints the total value to be saved.

TI 99/4A

## Bonus Program #5 SPORTS FORECASTER

The SPORTS FORECASTER can be a handy program if you enjoy sports. This program will take a team's current record and project, based on winning percentage, what the team's record will be at the end of the season.

```

10 CALL CLEAR
20 PRINT "SPORTS FORECASTER"
30 FOR Z=1 TO 500
40 NEXT Z
50 PRINT
60 PRINT "TO FORECAST A TEAM'S"
70 PRINT "FINAL WIN AND LOSS RECORD"
80 PRINT "BASED ON ITS CURRENT RECORD"
90 PRINT
100 PRINT "ANSWER THE QUESTION"
110 PRINT "THEN PRESS ENTER"
120 PRINT
130 PRINT "ENTER TOTAL GAMES TEAM PLAYS"
140 INPUT T
150 PRINT
160 PRINT "HOW MANY WINS DO THEY HAVE"
170 INPUT W
180 PRINT "HOW MANY LOSSES DO THEY HAVE"
190 INPUT L
200 P=W/(W+L)
210 Y=T*P
220 D=T-Y
230 PRINT "END OF THE SEASON PROJECTION"
240 PRINT
250 PRINT "WINS=";INT(Y);"LOSSES=";INT(D)+1

```

The VARIABLES are:

Z=time delay variable

T=total games in season

W=games won

L=games lost

P=percentage of games won

Y=end of year games projected won

D=end of year games projected lost

The forecast is completed in line 200 when the winning percentage (P) is established by dividing the number of completed games (W+L) into the total games won so far (W). The total wins for the year is estimated by multiplying the amount of games in the season (T) by the winning percentage (P). The year end losses are determined by subtracting the end of year projected games won (Y) from the total games in the season (T).

TI 99/4A

## Bonus Program #6 SHOWER MONITOR

Getting into the shower, day after day, and finding cold water can be a drag. I'm sure that large families know what I'm talking about. Bonus Program #7 has been designed to whip, morning bathroom confusion. It's called the SHOWER MONITOR. You type in the names and the computer picks the shower order.

```

10 RANDOMIZE
20 CALL CLEAR
30 PRINT "SHOWER MONITOR"
40 FOR T=1 TO 2000
50 NEXT T
60 CALL CLEAR
70 PRINT "THIS PROGRAM WILL HELP"
80 PRINT "DECIDE, IN A FAIR WAY"
90 PRINT "THE MORNING SHOWER ORDER"
100 PRINT
110 PRINT "TYPE ANSWER THEN ENTER"
120 PRINT "HOW MANY IN YOUR FAMILY"
130 INPUT P
140 PRINT
150 PRINT "TYPE IN THE NAMES, ONE AT"
160 PRINT "A TIME. THEN PRESS ENTER"
170 FOR H=1 TO P
180 INPUT N$(H)
190 NEXT H
200 CALL CLEAR
210 PRINT "THIS IS THE SHOWER ORDER:"
220 FOR R=1 TO P
230 X=INT(RND*P)+1
240 IF N$(X)="" THEN 230
250 PRINT N$(X)
260 N$(X)=""
270 NEXT R
280 GOTO 280

```

The VARIABLES are:

P=number of people in family  
H=array parking lot#  
X=random number  
N\$(X)=name of person X in array  
R=counting variable

You are probably wondering what an ARRAY is. An ARRAY is a computer parking lot. In an ARRAY you don't park cars. Rather, you park words and numbers. In this program we parked the name of each person in an ARRAY location (such as N\$(1)="Mom", N\$(2)="Larry", N\$(3)="Rick", etc.). The names are loaded into the ARRAY in lines 170-190. The RANDOM shower order is determined in lines 210-220. Can you figure out why a name isn't picked more than once?

TI 99/4A

## Bonus Program #7 ELECTION RETURN

Stage a mock primary with four candidates. Twenty-five precincts report, one at a time. Running totals are printed as each reports. When all the returns are in, the computer displays the final results.

```

10 CALL CLEAR
20 PRINT "ELECTION RETURNS"
30 FOR A=1 TO 1500
40 NEXT A
50 FOR X=1 TO 4
60 PRINT "INPUT NAME OF CANDIDATE #";X
70 INPUT N$(X)
80 NEXT X
90 CALL CLEAR
100 PRINT "THE POLLS JUST CLOSED"
110 PRINT "& THE RESULTS ARE COMING IN"
114 FOR Y=1 TO 500
118 NEXT Y
120 FOR P=1 TO 25
130 FOR X=1 TO 4
140 RANDOMIZE
150 N(X)=INT(RND*999)+1
160 NEXT X
170 CALL CLEAR
180 PRINT "PRECINCT #";P;"RESULTS"
190 FOR X=1 TO 4
200 PRINT N(X);"-";N$(X)
210 T(X)=N(X)+T(X)
220 NEXT X
230 FOR B=1 TO 1000
240 NEXT B
250 PRINT
260 PRINT "CURRENT TOTALS:"
270 FOR X=1 TO 4
280 PRINT T(X);"-";N$(X)
290 NEXT X
300 FOR C=1 TO 1000
310 NEXT C
320 NEXT P
330 CALL CLEAR
340 PRINT "ALL RETURNS IN, AND"
350 PRINT "THESE ARE THE TOTALS:"
360 FOR X=1 TO 4
370 PRINT T(X);"-";N$(X)
380 NEXT X

```

continued on next page.....

The VARIABLES are:

N\$(X)=candidates names 1-4

N(X)=votes per/precinct candidates 1-4

T(X)=votes total each candidate 1-4

A=time delay1

B=time delay2

C=time delay3

P=precinct #

X=arrays loading variable

In lines 50-80 the candidates' names are loaded into an array called N\$(X). The vote totals, for each precinct, are generated by line 150. The precinct totals and subtotals are printed out in lines 100-320. Lines 360-380 print the final results.

TI 99/4A

## Bonus Program #8 PRACTICAL JOKER

Are you ready for some laughs? If so, Bonus Program #8 is the one for you. It's called the PRACTICAL JOKE PROGRAM. Here's how it works:

You type the program into your computer while the victim isn't around. RUN the program. The computer will ask you questions about the victim. You answer all the questions until the computer says, "PRESS ANY KEY TO START THE JOKE". Press a key and the joke is ready for the victim. When the victim comes back, the computer will seem to know all about him/her. Just say that you're hooked up to the MASTER COMPUTER, and it knows EVERYTHING!!!

```

10 CALL CLEAR
20 PRINT "PRACTICAL JOKE PROGRAM"
30 FOR T = 1 TO 1000
40 NEXT T
50 CALL CLEAR
60 PRINT "WHAT IS THE VICTIM'S NAME";
70 INPUT N$
80 PRINT "IS THE VICTIM MALE OR FEMALE";
90 INPUT MF$
100 IF MF$="MALE" THEN 120 ELSE 110
110 IF MF$="M" THEN 120 ELSE 140
120 G$= "HE"
130 P$= "HIS"
140 IF MF$="FEMALE" THEN 160 ELSE 150
150 IF MF$="F" THEN 160 ELSE 180
160 G$= "SHE"
170 P$= "HER"
180 PRINT "WHAT CITY IS ";G$;" FROM ";
190 INPUT C$
200 PRINT "HOW OLD IS ";G$;" NOW";
210 INPUT A
220 PRINT "WHAT IS ";P$;" FAVORITE HOBBY";
230 INPUT H$
240 PRINT "WHAT IS ";P$;" NICKNAME";
250 INPUT NN$
260 PRINT "PRESS ENTER TO BEGIN JOKE.";
270 INPUT S$
280 CALL CLEAR
290 PRINT "PRESS ENTER KEY & I'LL TALK"
300 INPUT S$
310 PRINT "I CAN GUESS YOUR NAME"
320 PRINT "I'M THINKING..."
330 GOSUB 700
340 PRINT "YOU LOOK LIKE ";N$;
350 GOSUB 700
360 PRINT "I'LL JUST CALL YOU";
370 PRINT NN$
380 PRINT "IF THAT'S OKAY"
390 GOSUB 700

```

continued on next page.....

TI 99/4A

```

400 PRINT "THE TOUCH OF YOUR FINGERS"
410 PRINT
420 PRINT "TELLS ME YOU ARE"
430 PRINT
440 PRINT A;"YEARS OLD"
450 GOSUB 700
460 PRINT "YOU SMELL LIKE A PERSON FROM"
470 PRINT C$
480 PRINT
490 GOSUB 700
500 PRINT "COMPUTER CENTRAL TELLS ME"
510 PRINT "THAT YOU DIG"
520 PRINT H$
530 GOSUB 700
540 PRINT "NOW, YOU ASK ME A QUESTION"
550 PRINT "TYPE QUESTION, THEN ENTER"
560 INPUT Q$
570 GOSUB 700
580 PRINT "THAT'S TOO PERSONAL!!!"
590 PRINT "YOUR ACCESS HAS"
600 PRINT "BEEN TERMINATED UNTIL 1999"
610 END
700 FOR T=1 TO 3000
710 NEXT T
720 CALL CLEAR
730 RETURN

```

In the PRACTICAL JOKE program you are introduced to some new commands. Two are called GOSUB and RETURN (not the key). GOSUB 700 means, "goto the subroutine at 700". A SUBROUTINE is like a program, within a program. RETURN means "return to the main program". A SUBROUTINE always starts with GOSUB and ends with ENTER. This SUBROUTINE, beginning at line 700, causes a time delay and clears the screen. The command ELSE works with IF and THEN. If the condition isn't met in an IF/THEN statement, ELSE gives an alternate line number to GOTO.

VARIABLES are:

T=time delay  
 N\$=victim's name  
 MF\$=male or female  
 G\$=he or she  
 C\$=city  
 A=age  
 H\$=hobby  
 NN\$=nickname  
 P\$=his or her  
 Q\$=question  
 S\$=continue

TI 99/4a

## Bonus Program #9 NUMBER GAME FOR TWO

Computers are great for playing games. They can be programmed to make games, both unpredictable and exciting. Here is a super game for two people. The computer "pulls a number out of its hat", and the players take turns trying to guess the number. The player with the most correct guesses, after seven rounds, is the champ. Switch sides after seven rounds. You will be surprised at the strategies involved.

```

10 RANDOMIZE
20 CALL CLEAR
30 PRINT "WHAT IS PLAYER # 1'S NAME"
40 INPUT P1$
50 PRINT
60 PRINT "WHAT IS PLAYER # 2'S NAME"
70 INPUT P2$
80 R=R+1
90 N=INT(RND*500)+1
100 IF R>1 THEN 110 ELSE 130
110 PRINT "THE SCORE =";P1$;"=";A;" ";P2$;"=";B
120 IF R>7 THEN 460
130 FOR T=1 TO 1000
140 NEXT T
150 CALL CLEAR
160 PRINT "ROUND";R;"", ";P1$;" 'S TURN"
170 PRINT
180 PRINT "WHAT IS YOUR GUESS"
190 INPUT G1
200 IF G1<N THEN 210 ELSE 220
210 GOSUB 530
220 IF G1<N THEN 290
230 IF G1>N THEN 240 ELSE 250
240 GOSUB 550
250 IF G1>N THEN 290
260 PRINT "YOU GOT IT"
270 A=A+1
280 GOTO 80
290 FOR T=1 TO 1000
300 NEXT T
310 CALL CLEAR
320 PRINT "ROUND";R;"", ";P2$;" 'S TURN"
330 PRINT
340 PRINT "WHAT IS YOUR GUESS"
350 INPUT G2
360 IF G2<N THEN 370 ELSE 380
370 GOSUB 530
380 IF G2<N THEN 130
390 IF G2>N THEN 400 ELSE 410
400 GOSUB 550
410 IF G2>N THEN 130
420 PRINT "YOU GOT IT"
430 B=B+1
440 GOTO 80

```

continued on next page.....

TI 99/4A

```

450 CALL CLEAR
460 FOR T=1 TO 1000
470 NEXT T
480 IF A>B THEN 490 ELSE 510
490 PRINT P1$;" CREAMED ";P2$;" ";A;" TO ";B
500 END
510 PRINT P2$;" WASTED ";P1$;" ";B;" TO ";A
520 END
530 PRINT "TOO LOW"
540 RETURN
550 PRINT "TOO HIGH"
560 RETURN

```

The VARIABLES are:

P1\$=player #1  
 P2\$=player #2  
 A=player #1 score  
 B=player #2 score  
 G1=player #1 guess  
 G2=player #2 guess  
 R=round#  
 T=time delay variable  
 N=secret number

The secret number (N) is picked in line 90. To alter the limits of the secret number, you can change the 500 to a larger or smaller number. Player #1's turn occurs in lines 160-280. Player #2's turn takes place in 320-440. The final results are displayed in 450-500. In several places you may notice symbols like this: > or this: < . The symbol > means "greater than" and < means "less than". Can you make this game work with four players?

TI 99/4A

## Bonus Program #10 MATH SHARPENER

This program is called the MATH SHARPENER. I've been saving this program for the whiz kids. You must be one, or you wouldn't be reading this. The MATH SHARPENER has been designed to quiz both the beginner, and the advanced, on basic math skills. Don't use a scratch pad and I assure you, you'll get a work out. You will receive instructions when you RUN the program.

```

10 RANDOMIZE
20 S=0
30 CALL CLEAR
40 PRINT "MATH SHARPENER"
50 PRINT
60 PRINT "CHOOSE: 1)EASY 2)HARD"
70 PRINT "THEN PRESS ENTER"
80 INPUT L
90 PRINT "THE SYMBOLS ARE:"
100 PRINT "+ ADD      - SUBTRACT"
110 PRINT "* MULTIPLY  / DIVIDE"
120 PRINT
130 PRINT "EXAMPLES:"
140 PRINT "2+3=5      8-4=4"
150 PRINT "2*4=8      9/3=3"
160 PRINT
170 PRINT "PICK THE # OF THE FUNCTION"
180 PRINT "YOU WANT TO PRACTICE"
190 PRINT
200 PRINT "1)ADD"
210 PRINT "2)SUBTRACT"
220 PRINT "3)MULTIPLY"
230 PRINT "4)DIVIDE"
240 INPUT D
250 CALL CLEAR
260 PRINT "TYPE THE # OF QUESTIONS"
270 PRINT "YOU WANT. THEN PRESS ENTER"
280 INPUT Q
290 FOR T=1 TO Q
300 PRINT "ANSWER THE PROBLEM"
310 PRINT "THEN PRESS ENTER"
320 PRINT
330 IF L=1 THEN 340 ELSE 370
340 A=INT(RND*10)+1
350 B=INT(RND*10)+1
360 GOTO 390
370 A=INT(RND*200)+1
380 B=INT(RND*200)+1
390 IF D=2 THEN 400 ELSE 410
400 IF A<B THEN 330
410 IF D<>4 THEN 430

```

continued on next page.....

TI 99/4A

```

420 IF A<B THEN 330
430 IF D=4 THEN 440 ELSE 450
440 IF A/B <> INT(A/B) THEN 330
450 IF A=B THEN 330
460 IF B=1 THEN 330
470 IF D=1 THEN 480 ELSE 500
480 PRINT A;"+";B;"=";
490 INPUT C
500 IF D=2 THEN 510 ELSE 530
510 PRINT A;"-";B;"=";
520 INPUT C
530 IF D=3 THEN 540 ELSE 600
540 IF L=1 THEN 550 ELSE 570
550 PRINT A;"*";B;"=";
560 INPUT C
570 IF L=2 THEN 580 ELSE 600
580 PRINT INT(A/5);"*";INT(B/5);"=";
590 INPUT C
600 IF D=4 THEN 610 ELSE 630
610 PRINT A;"/";B;"=";
620 INPUT C
630 PRINT
640 IF D=1 THEN 650 ELSE 690
650 IF C=A+B THEN 660 ELSE 690
660 S=S+1
670 PRINT "ALRIGHT!"
680 GOTO 1060
690 IF D=2 THEN 700 ELSE 740
700 IF C=A-B THEN 710 ELSE 740
710 S=S+1
720 PRINT "RIGHT ON!"
730 GOTO 1060
740 IF D=3 THEN 750 ELSE 850
750 IF L=1 THEN 760 ELSE 800
760 IF C=A*B THEN 770 ELSE 850
770 S=S+1
780 PRINT "CORRECT!"
790 GOTO 1060
800 IF L=2 THEN 810 ELSE 850
810 IF C=INT(A/5)*INT(B/5) THEN 820 ELSE 850
820 S=S+1
830 PRINT "UNBELIEVABLE!"
840 GOTO 1060
850 IF D=4 THEN 860 ELSE 900
860 IF C=A/B THEN 870 ELSE 900
870 S=S+1
880 PRINT "YOWSAH!"
890 GOTO 1060
900 PRINT "YOU GOOFED/THE ANSWER IS ";
910 IF D=1 THEN 920 ELSE 940
920 PRINT A+B
930 GOTO 1060
940 IF D=2 THEN 950 ELSE 970

```

continued on next page.....

TI 99/4A

```

950 PRINT A-B
960 GOTO 1060
970 IF D=3 THEN 980 ELSE 1040
980 IF L=1 THEN 990 ELSE 1010
990 PRINT A*B
1000 GOTO 1060
1010 IF L=2 THEN 1020 ELSE 1040
1020 PRINT INT(A/5)*INT(B/5)
1030 GOTO 1060
1040 IF D=4 THEN 1050 ELSE 1060
1050 PRINT A/B
1060 FOR W=1 TO 1000
1070 NEXT W
1080 CALL CLEAR
1090 NEXT T
1100 PRINT "YOUR SCORE IS ";S
1110 PRINT " OUT OF ";Q;" RIGHT!"
1120 PRINT "PRESS ENTER TO CONTINUE"
1130 INPUT A$
1140 GOTO 10

```

The VARIABLES are:

L=easy/hard  
 D=function to practice  
 Q=# of questions  
 T=question loop  
 A=random number 1  
 B=random number 2  
 C=choice (your answer)  
 W=delay  
 S=score  
 A\$=continue

If you have problems getting this program to work properly, double check the lines mentioned in ERROR STATEMENTS. Retype the lines that look wrong. In this program, if D=1, then you are adding; if D=2, you are subtracting; D=3, means multiplying; and D=4 signifies division. If L=1, then the quizzes are easy. If L=2, then the quizzes are hard. Should you have the desire to make the program more difficult, you can increase the RANDOM NUMBERS, in lines 340-380.

## TIMEX

## Bonus Program #1 CARTOON ROBOT

Riddle: What's sweet, but square; high tech, yet down to earth; and brilliant, with the I.Q. of a doughnut?

Give up?

The answer is: The Fantabulous KISSING ROBOT

You'll key in a program using FOR and NEXT to make a cartoon. The commands FOR and NEXT are used for counting.

```

14 CLS
16 PRINT "()******()"
18 PRINT "***          ***"
20 PRINT "*** (O)      (O) ***"
22 PRINT "***          ***"
24 PRINT "***      V      ***"
26 PRINT "***          ***"
28 PRINT "***      ---     ***"
30 FOR T = 1 TO 10
31 NEXT T
32 CLS
34 PRINT "()******()"
36 PRINT "***          ***"
38 PRINT "*** (O)      (-) ***"
40 PRINT "***          ***"
42 PRINT "***      V      ***"
44 PRINT "***          ***"
46 PRINT "***      *      ***"
48 FOR T = 1 TO 10
49 NEXT T
50 CLS
52 PRINT "()******()"
54 PRINT "***          ***"
56 PRINT "*** (O)      (O) ***"
58 PRINT "***          ***"
60 PRINT "***      V      ***"
62 PRINT "***          ***"
64 PRINT "***      O      ***"
66 FOR T = 1 TO 10
67 NEXT T
68 GOTO 14

```

This program works like a real cartoon. The robot is printed on the screen, and erased, three times. Each time it is printed, there are small changes made, which give the illusion of movement. Line 68 GOTO 14 starts the entire process over again. The FOR/NEXT commands are used as time delays between pictures. You can change the speed of the cartoon by changing the 10, in the FOR/NEXT lines, to a different number. Decreasing the number will make the cartoon faster.

## TIMEX

## Bonus Program #2 FAMILY DECISION MAKER

How would you like to use your home computer for solving problems like, "Who will use the computer first, Jimmy or Bobby?"...or how about, "Should we use the t.v. to watch a movie, or play with the computer?". The FAMILY DECISION MAKER can help you solve these problems, and more. It will make the decision for you, by picking a random choice. All you have to do is to type in the options.

```

10 CLS
20 PRINT "FAMILY DECISION MAKER"
30 FOR T=1 TO 200
40 NEXT T
50 CLS
60 PRINT "TYPE IN THE OPTIONS AND"
70 PRINT "THE COMPUTER WILL DECIDE."
80 PRINT "WHAT IS OPTION NUMBER 1?"
90 INPUT A$
100 PRINT "WHAT IS OPTION NUMBER 2?"
110 INPUT B$
120 PRINT "I AM THINKING IT OVER....."
130 FOR T = 1 TO 175
140 NEXT T
150 LET C=INT(RND*2)+1
160 PRINT "MY CHOICE IS:"
170 IF C=2 THEN GOTO 200
180 PRINT A$
190 STOP
200 PRINT B$

```

(Hint: You must LIST long programs in sections. This is done by typing LIST followed by the LINE NUMBER you want to LIST from. Example: LIST 100. This would LIST lines 100 and above, until the screen is full.)

In this program the computer makes its choice in line 150. Lines 120 to 140 are where the computer is, "thinking it over". You probably noticed that the computer isn't really "thinking it over". It's actually counting up to 175, then executing line 160. The computer's choice is printed on the screen in lines 180-200.

The VARIABLES are:

T=time delay

C=choice

A\$=option NUMBER 1

B\$=option NUMBER 2

## TIMEX

## Bonus Program #3 M.P.G. RECORDER

If you're like me, you never take the trouble to figure out your car's miles per gallon (M.P.G.). Even having a calculator handy has never helped, though there are only three basic numbers to calculate. This is one instance in which wanting to use my computer motivates me to do the fairly simple task I've managed to ignore. My mechanic tells me that I should check my M.P.G. after every five fill-ups. That way, if my M.P.G. starts dropping, I can take my car in for a checkup...before it's too late.

```

10 CLS
20 PRINT "MPG CALCULATOR"
30 FOR T=1 TO 200
40 NEXT T
50 CLS
60 PRINT "THIS PROGRAM DETERMINES"
70 PRINT "THE MPG YOUR CAR GETS."
80 PRINT "HOW MANY MILES DID YOU DRIVE"
90 PRINT "DURING THE PAST 5 FILL-UPS?"
100 INPUT M
110 PRINT "HOW MANY GALLONS OF GAS DID YOU"
120 PRINT "USE IN THE PAST 5 FILL-UPS?"
130 INPUT G
140 LET MPG=M/G
150 PRINT "YOU HAVE BEEN GETTING ";MPG
160 PRINT "MILES PER GALLON"

```

Notice that we used G as the VARIABLE for gas, M as the VARIABLE for miles, and MPG as the VARIABLE for miles per gallon. In line 140,  $MPG=M/G$  means miles per gallon equals miles divided by gallons.

## TIMEX

## Bonus Program #4 COUPON CALCULATOR

Computers are pretty good at solving problems and presenting the results in a manner which is easy to read. This program can be used to display the amount of money you will save with your shopping coupons.

```

10 CLS
20 PRINT "COUPON CALCULATOR"
30 PRINT
40 PRINT "FIND OUT WHAT YOU WILL SAVE"
50 PRINT "WITH YOUR SHOPPING COUPONS"
60 PRINT "ANSWER THE QUESTION"
70 PRINT "THEN PRESS THE ENTER KEY."
80 PRINT
90 PRINT "HOW MANY COUPONS DO YOU HAVE?"
100 INPUT C
110 PRINT
120 PRINT "TO ENTER VALUE OF A COUPON"
130 PRINT "DO NOT USE A DOLLAR SIGN"
140 PRINT "DO USE A DECIMAL POINT"
150 PRINT
160 LET T=0
170 FOR E=1 TO C
180 PRINT "ENTER VALUE OF COUPON NUMBER ";E
190 INPUT A
200 LET T=T+A
210 NEXT E
220 PRINT
230 PRINT "$";T;" WILL BE SAVED"

```

The VARIABLES in this program are:

C=number of coupons

E=coupon#

A=value of coupons

T=total value of coupons

(Hint: See Program #2 if you forgot how to LIST a long program.)

Line 200 adds up the total, each time a value is entered into the computer. Line 230 prints the total value to be saved.

## TIMEX

## Bonus Program #5 SPORTS FORECASTER

The SPORTS FORECASTER can be a handy program if you enjoy sports. This program will take a team's current record and project, based on winning percentage, what the team's record will be at the end of the season.

```

10 CLS
20 PRINT "SPORTS FORECASTER"
30 FOR Z = 1 TO 200
40 NEXT Z
50 CLS
60 PRINT "TO FORECAST THE FINAL"
70 PRINT "WIN AND LOSS RECORD OF A TEAM"
80 PRINT "BASED ON CURRENT RECORD"
90 PRINT
100 PRINT "TYPE YOUR ANSWER"
110 PRINT "THEN PRESS ENTER"
120 PRINT
130 PRINT "ENTER TOTAL GAMES TEAM PLAYS"
140 INPUT T
150 PRINT
160 PRINT "HOW MANY WINS DO THEY HAVE?"
170 INPUT W
180 PRINT "HOW MANY LOSSES DO THEY HAVE?"
190 INPUT L
200 LET P = W/(W+L)
210 LET Y = T*P
220 LET D = T-Y
230 PRINT "END OF SEASON PROJECTION"
240 PRINT
250 PRINT "WINS=";INT(Y);" LOSSES=";INT(D)+1

```

The VARIABLES are:

Z=time delay variable

T=total games in season

W=games won

L=games lost

P=percentage of games won

Y=end of year games projected won

D=end of year games projected lost

The forecast is completed in line 200 when the winning percentage (P) is established by dividing the number of completed games (W+L) into the total games won so far (W). The total wins for the year is estimated by multiplying the amount of games in the season (T) by the winning percentage (P). The year end losses are determined by subtracting the end of year projected games won (Y) from the total games in the season (T).

## TIMEX

## Bonus Program #6 SHOWER MONITOR

Getting into the shower, day after day, and finding cold water can be a drag. I'm sure that large families know what I'm talking about. Bonus Program #6 has been designed to whip morning bathroom confusion. It's called the SHOWER MONITOR. You type in the names and the computer picks the shower order. (Hint: To make DIM press "D". DIM N\$(10,15) tells the computer that there will be a total of no more than 10 names with up to 15 letters each.)

```

10 DIM N$(10,15)
20 PRINT "SHOWER MONITOR"
30 FOR T=1 TO 250
40 NEXT T
50 CLS
60 PRINT "THIS PROGRAM WILL HELP"
70 PRINT "DECIDE, IN A FAIR WAY,"
80 PRINT "THE MORNING SHOWER ORDER"
90 PRINT
100 PRINT "TYPE ANSWER, THEN PRESS ENTER"
110 PRINT "HOW MANY IN YOUR FAMILY?"
120 INPUT P
130 PRINT
140 PRINT "TYPE IN THE NAMES, ONE AT"
150 PRINT "A TIME. THEN PRESS ENTER"
160 FOR H=1 TO P
170 INPUT N$(H)
180 NEXT H
190 CLS
200 PRINT "THIS IS THE SHOWER ORDER:"
210 FOR R=1 TO 20
220 LET H=INT(RND*P)+1
230 IF N$(H)=" " THEN GOTO 220
240 PRINT N$(H)
250 LET N$(H)=" "
260 NEXT R

```

The VARIABLES are:

P=number of people in family

H=array parking lot#

H=random number

N\$(H)=name of person H in array

R=counting variable

You are probably wondering what an ARRAY is. An ARRAY is a computer parking lot. In an ARRAY you don't park cars. Rather, you park words and numbers. In this program we parked the name of each person in an ARRAY location (such as N\$(1)="Mom", N\$(2)="Larry", N\$(3)="Rick", etc.). The names are loaded into the ARRAY in lines 160-180. The RANDOM shower order is determined in lines 200-260. Can you figure out why a name isn't picked more than once?

## TIMEX

## Bonus Program #7 ELECTION RETURNS

Stage a mock primary with four candidates. Twenty-five precincts report, one at a time. Running totals are printed as each reports. When all the returns are in, the computer displays the final results.

```

10 DIM N$(4,15)
20 DIM N(4)
30 DIM T(4)
40 PRINT "ELECTION RESULTS"
50 FOR T=1 TO 150
60 NEXT T
70 FOR X=1 TO 4
80 PRINT "INPUT NAME OF CANDIDATE NUMBER ";X
90 INPUT N$(X)
100 NEXT X
110 CLS
120 PRINT "THE POLLS JUST CLOSED,"
130 PRINT "AND THE RESULTS ARE COMING IN"
140 FOR T=1 TO 50
150 NEXT T
160 FOR P=1 TO 25
170 FOR X=1 TO 4
180 LET N(X)=INT(RND*999)+1
190 NEXT X
200 CLS
210 PRINT "PRECINCT NUMBER ";P;" RESULTS"
220 FOR X=1 TO 4
230 PRINT N(X); "-"; N$(X)
240 LET T(X)=N(X)+T(X)
250 NEXT X
260 FOR T=1 TO 50
270 NEXT T
280 PRINT
290 PRINT "CURRENT TOTALS:"
300 FOR X=1 TO 4
310 PRINT T(X); "-"; N$(X)
320 NEXT X
330 FOR C=1 TO 50
340 NEXT C
350 NEXT P
360 CLS
370 PRINT "ALL RETURNS IN, AND"
380 PRINT "THESE ARE THE TOTALS:"
390 FOR X=1 TO 4
400 PRINT T(X); "-"; N$(X)
410 NEXT X

```

A list of VARIABLES, and a program explanation, may be found on the next page.

## TIMEX

The VARIABLES are:

N\$(X)=candidates names 1-4

N(X)=votes per precinct candidates 1-4

T(X)=votes total each candidate 1-4

T=time delay

P=precinct #

X=arrays loading variable

In lines 70-100 the candidates' names are loaded into an array called N\$(X). The vote totals, for each precinct, are generated by line 180. The precinct totals and subtotals are printed out in lines 160-350. Lines 390-410 print the final results.

## TIMEX

## Bonus Program #8 PRACTICAL JOKER

Are you ready for some laughs? If so, Bonus Program #8 is the one for you. It's called the PRACTICAL JOKE PROGRAM. Here's how it works:

You type the program into your computer while the victim isn't around. RUN the program. The computer will ask you questions about the victim. You answer all the questions until the computer says, "PRESS ENTER TO START THE JOKE". Press ENTER and the joke is ready for the victim. When the victim comes back, the computer will seem to know all about him/her. Just say that you're hooked up to the MASTER COMPUTER, and it knows EVERYTHING!!! (Hints: Make GOSUB by pressing "H". Make RETURN by pressing "Y")

```

10 CLS
20 PRINT "PRACTICAL JOKE PROGRAM"
30 GOSUB 700
40 PRINT "ANSWER THE QUESTION"
50 PRINT "THEN PRESS ENTER."
60 PRINT "WHAT IS THE NAME OF THE VICTIM?"
70 INPUT N$
80 PRINT "IS THE VICTIM MALE OR FEMALE?"
90 INPUT M$
100 IF M$ = "MALE" THEN GOTO 120
110 IF M$ = "M" THEN GOTO 120
115 GOTO 140
120 LET G$ = "HE"
130 LET P$ = "HIS"
135 IF G$ = "HE" THEN GOTO 180
140 IF M$ = "FEMALE" THEN GOTO 160
150 IF M$ = "F" THEN GOTO 160
155 GOTO 80
160 LET G$ = "SHE"
170 LET P$ = "HER"
180 PRINT "WHAT CITY IS ";G$;" FROM?"
190 INPUT C$
200 PRINT "HOW OLD IS ";G$
210 INPUT A
220 PRINT "WHAT IS ";P$;" FAVORITE HOBBY?"
230 INPUT H$
240 PRINT "WHAT IS ";P$;" NICKNAME?"
250 INPUT O$
260 PRINT "PRESS ENTER TO START JOKE.";
270 INPUT S$
280 CLS
290 PRINT "PRESS ENTER SO I CAN TALK TO YOU"
300 INPUT S$
310 PRINT "I CAN GUESS YOUR NAME"
320 PRINT "I AM THINKING..."
330 GOSUB 700
340 PRINT "YOU LOOK LIKE ";N$
350 GOSUB 700

```

continued on next page.....

## TIMEX

```

360 PRINT "I WILL JUST CALL YOU"
370 PRINT O$
380 PRINT "IF THAT IS OK"
390 GOSUB 700
400 PRINT "THE TOUCH OF YOUR FINGERS"
410 PRINT "TELLS ME THAT YOU ARE"
420 PRINT "AT LEAST"
430 PRINT A;" YEARS OLD"
440 GOSUB 700
450 PRINT "YOU SMELL LIKE A"
460 PRINT "PROGRAMMER I MET FROM"
470 PRINT C$
480 GOSUB 700
490 PRINT "THE CENTRAL COMPUTER"
500 PRINT "TELLS ME THAT YOU LIKE"
510 PRINT
520 PRINT H$
530 GOSUB 700
540 PRINT "NOW, YOU ASK ME A QUESTION"
550 PRINT "TYPE YOUR QUESTION"
560 PRINT "THEN PRESS ENTER"
570 INPUT Q$
580 GOSUB 700
590 PRINT "THAT IS TOO PERSONAL"
600 PRINT O$
610 PRINT "YOUR ACCESS TO THE"
620 PRINT "MASTER COMPUTER"
630 PRINT "HAS BEEN TERMINATED"
640 PRINT "UNTIL 1999"
700 FOR T=1 TO 75
710 NEXT T
720 CLS
730 RETURN

```

In the PRACTICAL JOKE program you are introduced to some new commands. Two are called GOSUB and RETURN (not the key). GOSUB 700 means, "goto the subroutine at 700". A SUBROUTINE is like a program, within a program. RETURN means "return to the main program". A SUBROUTINE always starts with GOSUB and ends with ENTER. This SUBROUTINE, beginning at line 700, causes a time delay and clears the screen.

## VARIABLES are:

T=time delay  
 N\$=victim's name  
 M\$=male or female  
 G\$=he or she  
 C\$=city  
 A=age  
 H\$=hobby  
 O\$=nickname  
 P\$=his or her  
 Q\$=question  
 S\$=continue

## TIMEX

## Bonus Program #9 NUMBER GAME FOR TWO

Computers are great for playing games. They can be programmed to make games, both unpredictable and exciting. Here is a super game for two people. The computer "pulls a number out of its hat", and the players take turns trying to guess the number. The player with the most correct guesses, after seven rounds, is the champ. Switch sides after seven rounds. You will be surprised at the strategies involved.

```

10 PRINT "WHAT IS THE NAME"
20 PRINT "OF PLAYER NUMBER 1?"
30 INPUT A$
40 PRINT "WHAT IS THE NAME"
50 PRINT "OF PLAYER NUMBER 2?"
60 INPUT B$
70 LET R=0
80 LET A=0
90 LET B=0
100 LET N=INT(RND*100)+1
110 LET R=R+1
120 PRINT "THE SCORE IS: ";A$;"=";A;" ";B$;"=";B
130 IF R>7 THEN GOTO 460
140 FOR T=1 TO 75
150 NEXT T
160 CLS
170 PRINT "ROUND ";R;"", ";A$;" S TURN"
180 PRINT
190 PRINT "WHAT IS YOUR GUESS?"
200 INPUT Y
210 IF Y=N THEN GOTO 270
220 IF Y>N THEN GOTO 250
230 GOSUB 540
240 IF Y<N THEN GOTO 300
250 GOSUB 560
260 IF Y>N THEN GOTO 300
270 PRINT "YOU GOT IT."
280 LET A=A+1
290 GOTO 100
300 FOR T=1 TO 75
310 NEXT T
320 CLS
330 PRINT "ROUND ";R;"", ";B$;" S TURN"
340 PRINT
350 PRINT "WHAT IS YOUR GUESS?"
360 INPUT Z
370 IF Z=N THEN GOTO 430
380 IF Z>N THEN GOTO 410
390 GOSUB 540

```

continued on next page.....

## TIMEX

```

400 IF Z<N THEN GOTO 140
410 GOSUB 560
420 IF Z>N THEN GOTO 140
430 PRINT "YOU GOT IT."
440 LET B=B+1
450 GOTO 100
460 CLS
470 FOR T=1 TO 75
480 NEXT T
490 IF B>A THEN GOTO 520
500 PRINT A$;" CREAMED ";B$;" ";A;" TO ";B
510 STOP
520 PRINT B$;" WASTED ";A$;" ";B;" TO ";A
530 STOP
540 PRINT "TOO LOW"
550 RETURN
560 PRINT "TOO HIGH"
570 RETURN

```

The VARIABLES are:

```

A$=player #1
B$=player #2
A=player #1 score
B=player #2 score
Y=player #1 guess
Z=player #2 guess
R=round number
T=time delay variable
N=secret number

```

The secret number (N) is picked in line 100. To alter the limits of the secret number, you can try changing the 100 to a larger or smaller number. In several places you may notice symbols like this: > or this: < . The symbol > means "greater than" and < means "less than". Can you make this game work with four players?

## TIMEX

## Bonus Program #10 MATH SHARPENER

This program is called the MATH SHARPENER. It has been designed to help improve your math skills. It is set up for multiplication, but can easily be changed to work with addition and subtraction. Don't use a scratch pad and I assure you, you'll get a work out.

```

10 PRINT "THIS IS A PROGRAM"
20 PRINT "TO SHARPEN YOUR MATH"
30 PRINT
40 PRINT "HOW MANY QUESTIONS DO YOU WANT?"
50 INPUT Q
60 CLS
70 LET S=0
80 FOR T=1 TO Q
90 LET A= INT(RND*20)+1
100 LET B= INT(RND*20)+1
110 CLS
120 PRINT "ANSWER THE QUESTION"
130 PRINT "THEN PRESS ENTER"
140 PRINT
150 PRINT A;"*";B;"=?"
160 INPUT C
170 PRINT
180 IF C=A*B THEN GOTO 260
190 PRINT
200 PRINT "YOU GOOFED, THE ANSWER IS: ";
210 PRINT A*B
220 FOR X=1 TO 75
230 NEXT X
240 NEXT T
250 GOTO 310
260 PRINT "YOU GOT IT"
270 FOR X=1 TO 75
280 NEXT X
290 LET S=S+1
300 NEXT T
310 CLS
320 PRINT "SCORE= ";S;"-RIGHT ";Q-S;"-WRONG"
330 PRINT
340 PRINT "TO PLAY AGAIN"
350 PRINT "PRESS THE ENTER KEY"
360 INPUT F$
370 CLS
380 GOTO 40

```

A list of VARIABLES and program information are on the next page.....

## TIMEX

The VARIABLES are:

Q=number of questions

S=score

T=question loop

A=random number 1

B=random number 2

C=response

X=delay variable

P\$=play again

Lines 90 and 100 are where the RANDOM NUMBERS for the questions are decided. Increase these numbers to make the quiz harder. To make the quiz work with addition, you must change all the multiplication symbols (\*) to addition symbols (+). Subtraction can be accomplished in this fashion also.

## TRS-80 and RADIO SHACK COLOR

## Bonus Program #1 CARTOON ROBOT

Riddle: What's sweet, but square; high tech, yet down to earth; and brilliant, with the I.Q. of a doughnut?

Give up?

The answer is: The fabulous KISSING ROBOT

You'll key in a program using FOR and NEXT to make a cartoon. The commands FOR and NEXT are used for counting.

```

14 CLS
16 PRINT "()******()"
18 PRINT "**          **"
20 PRINT "** (O)      (O) **"
22 PRINT "**          **"
24 PRINT "**      V      **"
26 PRINT "**          **"
28 PRINT "**      ---      **"
30 FOR T = 1 TO 75: NEXT T
32 CLS
34 PRINT "()******()"
36 PRINT "**          **"
38 PRINT "** (O)      (-) **"
40 PRINT "**          **"
42 PRINT "**      V      **"
44 PRINT "**          **"
46 PRINT "**      O      **"
48 FOR T = 1 TO 75: NEXT T
50 CLS
52 PRINT "()******()"
54 PRINT "**          **"
56 PRINT "** (O)      (O) **"
58 PRINT "**          **"
60 PRINT "**      V      **"
62 PRINT "**          **"
64 PRINT "**      O      **"
66 FOR T = 1 TO 75: NEXT T
68 GOTO 14

```

This program works like a real cartoon. The robot is printed on the screen, and erased, three times. Each time it is printed, there are small changes made, which give the illusion of movement. Line 68 GOTO 14 starts the entire process over again. The FOR/NEXT commands are used as time delays between pictures. You can change the speed of the cartoon by changing the 75, in the FOR/NEXT lines, to a different number. Decreasing the number will make the cartoon faster.

TRS-80

## Bonus Program #2 FAMILY DECISION MAKER

How would you like to use your home computer for solving problems like, "Who will use the computer first, Jimmy or Bobby?"...or how about, "Should we use the t.v. to watch a movie, or play with the computer?". The FAMILY DECISION MAKER can help you solve these problems, and more. It will make the decision for you, by picking a random choice. All you have to do is to type in the options.

```

10 CLS
20 PRINT "FAMILY DECISION MAKER"
30 FOR T = 1 TO 1500:NEXT T
40 CLS
50 PRINT "TYPE IN THE OPTIONS"
60 PRINT "AND THE COMPUTER WILL DECIDE"
70 INPUT "WHAT IS OPTION #1";O1$
80 INPUT "WHAT IS OPTION #2";O2$
90 PRINT "I'M THINKING IT OVER....."
100 FOR T = 1 TO 3000:NEXT T
110 C=RND(2)
120 CLS
130 PRINT "MY CHOICE IS:"
140 IF C = 1 THEN PRINT O1$
150 IF C = 2 THEN PRINT O2$

```

In this program the computer makes its choice in line 110. Lines 90 and 100 are where the computer is, "thinking it over". You probably noticed that the computer isn't really "thinking it over". It's actually counting up to 3000, then executing line 110. The choice is printed on the screen in lines 130-150.

The VARIABLES are:

T=time delay

C=choice

O1\$=option #1

O2\$=option #2

TRS-80

## Bonus Program #3 REACTION TIMER

Here is a program to test your reaction time. When the computer says "GO!", you must press the BREAK key as quickly as you can. Compare your score with the chart in the program. Good luck!

```

10 CLS
20 PRINT "TEST YOUR REACTION TIME"
30 PRINT "AGAINST THE COMPUTER."
40 PRINT "WHEN THE COMPUTER SAYS 'GO!'"
50 PRINT "PRESS THE BREAK KEY"
60 PRINT "YOUR SCORE IS THE HIGHEST NUMBER YOU SEE"
70 PRINT:PRINT "01-10=LIGHTNIN' 10-20=QUICK!"
80 PRINT "20-30=AVERAGE 30-50=NAPPING"
90 PRINT:PRINT:PRINT "PRESS ENTER"
100 PRINT "WHEN YOU ARE READY"
110 INPUT A$
120 CLS:PRINT "ON YOUR MARK"
130 FOR T = 1 TO 1000:NEXT T:PRINT "GET SET!"
140 FOR T = 1 TO RND(5000):NEXT T
150 CLS:PRINT "GO!"
160 FOR T = 1 TO 50:PRINT T:NEXT T
170 PRINT "SOMEONE WAKE THIS PERSON UP!"

```

The VARIABLE, of the FOR/NEXT statement in line 140, equals a RANDOM INTEGER between one and five thousand. This causes the time delay to be different each time the program is RUN. When you press the BREAK key the computer will say, "Break in 160". This is normal for the program. Your score is the highest number you see. Type RUN and press ENTER to play again.

TRS-80

## Bonus Program #4 M.P.G. RECORDER

If you're like me, you never take the trouble to figure out your car's miles per gallon (M.P.G.). Even having a calculator handy has never helped, though there are only three basic numbers to calculate. This is one more instance in which wanting to use my computer motivates me to do the fairly simple task I've managed to ignore. My mechanic tells me that I should check my M.P.G. after every five fill-ups. That way, if my M.P.G. starts dropping, I can take my car in for a checkup...before it's too late.

```

10 CLS
20 PRINT "MPG CALCULATOR"
30 FOR T = 1 TO 1500:NEXT T
40 CLS
50 PRINT "THIS IS A PROGRAM TO FIGURE OUT"
60 PRINT "THE MILES PER GALLON YOUR CAR GETS"
70 PRINT "HOW MANY MILES HAVE YOU DRIVEN"
80 PRINT "DURING THE PAST FIVE FILL-UPS"
90 INPUT M
100 PRINT "HOW MANY GALLONS OF GAS DID YOU USE"
110 PRINT "IN THE PAST FIVE FILL-UPS"
120 INPUT G
130 MPG=M/G
140 PRINT "YOU HAVE BEEN GETTING "MPG
150 PRINT "MILES PER GALLON"

```

Notice that we used G as the VARIABLE for gas, M as the VARIABLE for miles, and MPG as the VARIABLE for miles per gallon. In line 130,  $MPG=M/G$  means miles per gallon equals miles divided by gallons.

TRS-80

## Bonus Program #5 COUPON CALCULATOR

Computers are pretty good at solving problems and presenting the results in a manner which is easy to read. This program can be used to display the amount of money you will save with your shopping coupons.

```

10 CLS
20 PRINT "COUPON CALCULATOR"
30 PRINT:PRINT "TO FIND OUT HOW MUCH YOU'LL SAVE"
40 PRINT "WITH YOUR SHOPPING COUPONS"
50 PRINT "ANSWER THE FOLLOWING QUESTIONS"
60 PRINT "THEN PRESS THE ENTER KEY"
70 PRINT:PRINT "HOW MANY COUPONS DO YOU HAVE"
80 INPUT C
90 PRINT:PRINT "ENTER THE AMOUNT OF A COUPON"
100 PRINT "DON'T USE A DOLLAR SIGN"
110 PRINT "DO USE A DECIMAL POINT":PRINT
120 FOR E = 1 TO C: PRINT "ENTER VALUE OF COUPON #"E
130 INPUT A
140 T=T+A
150 NEXT E
160 PRINT:PRINT "$"T"WILL BE SAVED"

```

The VARIABLES in this program are:

C=number of coupons

E=coupon#

A=value of coupons

T=total value of coupons

Line 140 adds up the total, each time a value is entered into the computer. Line 160 prints the total value to be saved.

TRS-80

## Bonus Program #6 SPORTS FORECASTER

The SPORTS FORECASTER can be a handy program if you enjoy sports. This program will take a team's current record and project, based on winning percentage, what the team's record will be at the end of the season.

```

10 CLS
20 PRINT "SPORTS FORECASTER"
30 FOR Z = 1 TO 1500:NEXT Z:PRINT
40 PRINT "THIS PROGRAM WILL FORECAST A TEAM'S"
50 PRINT "FINAL WIN AND LOSS RECORD"
60 PRINT "BASED ON ITS CURRENT RECORD"
70 PRINT:PRINT "ANSWER EACH QUESTION"
80 PRINT "THEN PRESS ENTER"
90 PRINT:PRINT "HOW MANY GAMES DOES THE TEAM PLAY"
100 INPUT T
110 PRINT:PRINT "HOW MANY WINS DO THEY HAVE NOW"
120 INPUT W
130 PRINT "HOW MANY LOSSES DO THEY HAVE NOW"
140 INPUT L
150 P=W/(W+L):Y=T*P:D=T-Y
160 PRINT:PRINT "END OF THE SEASON PROJECTION:"
170 PRINT:PRINT "WINS="INT(Y)"LOSSES="INT(D)+1

```

The VARIABLES are:

Z=time delay variable

T=total games in season

W=games won

L=games lost

P=percentage of games won

Y=end of year games projected won

D=end of year games projected lost

The forecast is completed in line 150 when the winning percentage (P) is established by dividing the number of completed games (W+L) into the total games won so far (W). The total wins for the year is estimated by multiplying the amount of games in the season (T) by the winning percentage (P). The year end losses are determined by subtracting the end of year projected games won (Y) from the total games in the season (T).

TRS-80

## Bonus Program #7 SHOWER MONITOR

Getting into the shower, day after day, and finding cold water can be a drag. I'm sure that large families know what I'm talking about. Bonus Program #7 has been designed to whip, morning bathroom confusion. It's called the SHOWER MONITOR. You type in the names and the computer picks the shower order.

```

10 CLS
20 PRINT "SHOWER MONITOR"
30 FOR T= 1 TO 2000:NEXT T:CLS
40 PRINT "THIS PROGRAM IS DESIGNED TO HELP"
50 PRINT "FAMILIES DECIDE, IN A FAIR WAY"
60 PRINT "THE ORDER IN WHICH THE SHOWER IS USED"
70 PRINT "IN THE MORNING."
80 PRINT:PRINT "EACH PERSON'S NAME IS TYPED INTO THE"
90 PRINT "COMPUTER. THEN THE COMPUTER RANDOMLY"
100 PRINT "CHOOSES THE ORDER (AS IF OUT OF A HAT)."
```

110 PRINT "TYPE EACH ANSWER, THEN PRESS ENTER."

120 PRINT:PRINT "HOW MANY PEOPLE IN YOUR FAMILY"

130 INPUT P

140 PRINT:PRINT "TYPE IN THE NAMES, ONE AT A TIME."

150 PRINT "THEN PRESS ENTER."

160 FOR H = 1 TO P

170 INPUT N\$(H)

180 NEXT H

190 CLS:PRINT "THIS IS THE SHOWER ORDER TODAY:"

200 PRINT:FOR R = 1 TO P

210 X=RND(P)

220 IF N\$(X)="" THEN 210

230 PRINT N\$(X)

240 N\$(X)=""

250 NEXT R

260 GOTO 260

The VARIABLES are:

P=number of people in family

H=array parking lot#

X=random number

N\$(X)=name of person X in array

R=counting variable

You are probably wondering what an ARRAY is. An ARRAY is a computer parking lot. In an ARRAY you don't park cars. Rather, you park words and numbers. In this program we parked the name of each person in an ARRAY location (such as N\$(1)="Mom",N\$(2)="Larry",N\$(3)="Rick",etc). The names are loaded into the ARRAY in lines 160-180. The RANDOM shower order is determined in lines 200-250. Can you figure out why a name isn't picked more than once?

TRS-80

## Bonus Program #8 NUMBER GAME FOR TWO

Computers are great for playing games. They can be programmed to make games, both unpredictable and exciting. Here is a super game for two people. The computer "pulls a number out of its hat", and the players take turns trying to guess the number. The player with the most correct guesses, after seven rounds, is the champ. Switch sides after seven rounds. You will be surprised at the strategies involved.

```

10 CLS
20 PRINT "THIS IS A NUMBER GAME FOR TWO PEOPLE"
30 PRINT "THE COMPUTER PICKS A NUMBER BETWEEN 1 AND 500."
40 PRINT "THE PLAYERS TAKE TURNS GUESSING THE NUMBER"
50 PRINT "UNTIL SOMEONE GUESSES THE NUMBER"
60 PRINT "THE PLAYER GUESSING THE MOST NUMBERS,"
70 PRINT "AFTER 7 ROUNDS, IS THE WINNER"
80 PRINT : INPUT "WHAT IS PLAYER #1'S NAME ";P1$
90 PRINT : INPUT "WHAT IS PLAYER #2'S NAME ";P2$
100 R=R+1:N = RND(500)
110 IF R>1 THEN PRINT"THE SCORE IS "P1$="P1" "P2$="P2
120 FOR T = 1 TO 2500:NEXT T
130 IF R > 7 THEN 260
140 FOR T = 1 TO 1000: NEXT T
150 CLS : PRINT "ROUND "R", "P1$"'S TURN"
160 PRINT : INPUT "WHAT IS YOUR GUESS ";G1
170 IF G1 < N THEN PRINT "TOO LOW, "P1$:GOTO 200
180 IF G1 > N THEN PRINT "TOO HIGH":GO TO 200
190 PRINT "YOU GOT IT "P1$:P1 = P1 + 1:GOTO 100
200 FOR T = 1 TO 1000: NEXT T
210 CLS:PRINT "ROUND "R", "P2$"'S TURN"
220 PRINT:INPUT "WHAT IS YOUR GUESS ";G2
230 IF G2 < N THEN PRINT "TOO LOW":GOTO 140
240 IF G2 > N THEN PRINT "TOO HIGH":GOTO 140
250 PRINT "YOU GOT IT "P2$:P2 = P2 +1:GOTO 100
260 CLS:FOR T = 1 TO 1000: NEXT T
270 IF P1 > P2 THEN PRINT P1$" CREAMED "P2$" "P1" TO "PW:END
280 PRINT P2$" WASTED "P1$" "P2" TO "P1

```

The VARIABLES are:

P1\$=player #1

P2\$=player #2

P1=player #1 score

P2=player #2 score

G1=player #1 guess

G2=player #2 guess

R=round#

T=time delay variable

N=secret number

The secret number (N) is picked in line 100. To alter the limits of the secret number, you can change the 500 to a larger or smaller number. Try 10000, for instance. In several places you may notice symbols like this: > or this: < . The symbol > means "greater than" and < means "less than". Can you make this game work with four players?

TRS-80

## Bonus Program #9 PRACTICAL JOKER

Are you ready for some laughs? If so, Bonus Program #9 is the one for you. It's called the PRACTICAL JOKE PROGRAM. Here's how it works:

You type the program into your computer while the victim isn't around. RUN the program. The computer will ask you questions about the victim. You answer all the questions until the computer says, "PRESS ANY KEY TO START THE JOKE". Press a key and the joke is ready for the victim. When the victim comes back, the computer will seem to know all about him/her. Just say that you're hooked up to the MASTER COMPUTER, and it knows EVERYTHING!!!

```

10 CLS
20 PRINT "PRACTICAL JOKE PROGRAM"
30 FOR T=1 TO 1000:NEXT T
40 CLS
50 INPUT "WHAT IS THE VICTIM'S NAME";N$
60 INPUT "IS THE VICTIM MALE OR FEMALE";MF$
70 IF MF$= "MALE" OR MF$= "M" THEN G$= "HE":P$= "HIS"
80 IF MF$= "FEMALE" OR MF$= "F" THEN G$= "SHE":P$= "HER"
90 PRINT "WHAT CITY IS "G$" FROM?":INPUT C$
100 PRINT "HOW OLD IS "G$" NOW?":INPUT A
110 PRINT "WHAT IS "P$" FAVORITE HOBBY":INPUT H$
120 PRINT "WHAT IS "P$" NICKNAME":INPUT NN$
130 PRINT "PRESS ANY KEY TO START THE JOKE."
140 S$=INKEY$:IF S$=""THEN 140
150 CLS
160 PRINT "PRESS A KEY AND I WILL TALK TO YOU"
170 S$=INKEY$:IF S$=""THEN 170
180 PRINT "HELLO, LET ME TRY TO GUESS YOUR NAME."
190 PRINT "I'M THINKING..."
200 GOSUB 400
210 PRINT "YOU LOOK LIKE SOME TYPE OF "N$:GOSUB400
220 PRINT "BUT I HOPE YOU WON'T MIND IF I CALL YOU"
230 PRINT NN$:GOSUB400
240 PRINT "THE TOUCH OF YOUR FINGERS":PRINT
250 PRINT "LEADS ME TO BELIEVE YOU ARE AT LEAST":PRINT
260 PRINT A" YEARS OLD":GOSUB400
270 PRINT "AND YOU SMELL LIKE A PERSON FROM":PRINT
280 PRINT C$:GOSUB400
290 PRINT "THE CENTRAL COMPUTER TELLS ME YOU LIKE:":PRINT
300 PRINT H$:GOSUB400
310 PRINT "NOW, IT'S YOUR TURN TO ASK ME A QUESTION"
320 INPUT "TYPE YOUR QUESTION, THEN PRESS ENTER";Q$
330 GOSUB400
340 PRINT "SORRY, "NN$" THAT'S TOO PERSONAL!"
350 PRINT "YOUR ACCESS TO THE MASTER COMPUTER HAS"
360 PRINT "BEEN TERMINATED UNTIL 1999":END
400 FOR T = 1 TO 4000:NEXTT:CLS:RETURN

```

continued on next page....

## TRS-80

In the PRACTICAL JOKE program you are introduced to some new commands. Two are called GOSUB and RETURN (not the key). GOSUB 400 means, "goto the subroutine at 400". A SUBROUTINE is like a program, within a program. RETURN means "return to the main program". A SUBROUTINE always starts with GOSUB and ends with RETURN. This SUBROUTINE, line 400, causes a time delay and clears the screen. In lines 140 and 170 is another, new command, called INKEY. INKEY\$ tells the computer to wait for a key to be pressed. In line 140, if no key is pressed, the computer waits at line 140. When a key is pressed, the program proceeds on to line 150.

## VARIABLES are:

T=time delay  
N\$=victim's name  
MF\$=male or female  
G\$=he or she  
C\$=city  
A=age  
H\$=hobby  
NN\$=nickname  
P\$=his or her  
Q\$=question  
S\$=inkey\$

TRS-80

## Bonus Program #10 MATH SHARPENER

This program is called MATH SHARPENER. I've been saving this program for the whiz kids. You must be one, or you wouldn't be reading this. The MATH SHARPENER has been designed to quiz both the beginner and the advanced on basic math skills. Don't use a scratch pad and I assure you, you'll get a work out. You will receive instructions when you RUN the program.

```

10 REM***MATH SHARPENER***
20 S=0:CLS
30 PRINT"THIS IS A PROGRAM TO SHARPEN YOUR MATH"
40 PRINT
50 PRINT "PICK#: 1)EASY 2)HARD, THEN ENTER"
60 INPUT L
70 PRINT"THE SYMBOLS ARE:"
80 PRINT"+ ADD          - SUBTRACT"
90 PRINT"* MULTIPLY     / DIVIDE":PRINT
100 PRINT"EXAMPLES:"
110 PRINT"2+3=5          8-4=4"
120 PRINT"2*4=8          9/3=3":PRINT
130 PRINT"PICK THE NUMBER OF THE FUNCTION"
140 PRINT"YOU WANT TO PRACTICE:":PRINT
150 PRINT"1)ADD"
160 PRINT"2)SUBTRACT"
170 PRINT"3)MULTIPLY"
180 PRINT"4)DIVIDE"
190 INPUT D
200 CLS:PRINT"TYPE THE NUMBER OF QUESTIONS"
210 INPUT"YOU WANT, THEN PRESS ENTER":Q
220 FOR T=1 TO Q
230 PRINT"ANSWER THE PROBLEM, THEN PRESS ENTER":PRINT
240 IF L=1 THEN A=RND(10):B=RND(10)
250 IF L=2 THEN A=RND(200):B=RND(200)
260 IF D=2 AND A<B THEN 240
270 IF D=4 AND A=0 OR B=0 THEN 240
280 IF D=4 AND A<B THEN 240
290 IF D=4 AND A/B <> INT(A/B) THEN 240
300 IF A=B OR B=1 THEN 240
310 IF D=1 THEN PRINT A"+"B="";:INPUTC
320 IF D=2 THEN PRINT A "-"B="";:INPUTC
330 IF D=3 AND L=1 THEN PRINT A*"B="";:INPUT C
340 IF D=3 AND L=2 THEN PRINT INT(A/5)*"INT(B/5)"="";:INPUTC
350 IF D=4 THEN PRINT A/"B="";:INPUTC

```

continued on next page...

## TRS-80

```

360 IF D=1 AND C=A+B THEN S=S+1:PRINT "ALRIGHT!":GOTO490
370 IF D=2 AND C=A-B THEN S=S+1:PRINT"RIGHT ON!":GOTO490
380 IF D=3 AND L=1 AND C=A*B THEN S=S+1:Z=1
390 IF Z=1 THEN PRINT "CORRECT!":Z=0:GOTO490
400 IF D=3 AND L=2 AND C=INT(A/5)*INT(B/5) THEN S=S+1:Z=2
410 IF Z=2 THEN PRINT "GREAT!":Z=0:GOTO490
420 IF D=4 AND C=A/B THEN S=S+1:PRINT"YOWSAH!":GOTO490
430 PRINT"YOU GOOFED. THE ANSWER WAS:";
440 IF D=1 THEN PRINT A+B
450 IF D=2 THEN PRINT A-B
460 IF D=3 AND L=1 THEN PRINT A*B
470 IF D=3 AND L=2 THEN PRINT INT(A/5)*INT(B/5)
480 IF D=4 THEN PRINT A/B
490 FOR W=1TO800:NEXT W:CLS:NEXT T
500 PRINT"YOUR SCORE IS:"S" OUT OF "Q" RIGHT!"
510 PRINT"PRESS ANY KEY TO CONTINUE"
520 A$=INKEY$:IF A$=""THEN 520
530 GOTO 10

```

The VARIABLES are:

L=easy/hard

D=function to practice

Q=# of questions

T=question loop

A=random number 1

B=random number 2

C=choice (your answer)


W=delay

S=score

A\$=get variable

Z=correct multiplication flag

If you have problems getting this program to work properly, double check lines 240-500. These are the lines where most of the mathematical processing takes place. In this section, if D=1, then you are adding; if D=2, you are subtracting; D=3, means multiplying; and D=4 signifies division. If L=1, then the quizzes are easy. If L=2, then the quizzes are hard. Should you have the desire to make the program more difficult, you can increase the RANDOM NUMBERS, in lines 240-250.



0-590-39012-0



SCHOLASTIC INC.

730 BROADWAY, NEW YORK, NEW YORK 10003